


# HEWLETT-PACKARD

Step-by-Step Solutions  
For Your HP Calculator

## Engineering Applications

$$\begin{aligned}
 &= \sum_{i=1}^{n_1} x_i \frac{1}{n_1} S_{\bar{D}} = 10 \text{ cm} + \frac{k+1}{2} y_i \\
 A^* &= \frac{1}{M} \left[ \left( \frac{n_1}{k+1} \right) \left( 1 + \frac{k-1}{2} M^2 \right) \right] \\
 D_i &= x_i - y_i \quad 2\pi c_1 \quad \sigma T^4 \quad x \quad s_1 = \frac{x_x}{c_3} \\
 &= \sum_{k=1}^n \text{LAT}_k \left( \frac{1}{2} \text{DEP}_k + \sum_{j=1}^k \text{DEP}_j \right) \frac{S_D}{\sqrt{n}} \\
 \bar{D} &= \frac{1}{n} \sum_{i=1}^n D_i \quad \text{Area} \quad \bar{y} \quad \sqrt{\left( \frac{s_x - s_y}{2} \right)^2 + \frac{F}{mg}} \\
 E_{b\lambda} &= \frac{2\pi c_1}{\lambda^5 (e^{c_2/\lambda T} - 1)} \quad k = \frac{F}{x} = \frac{mg}{x} = 0
 \end{aligned}$$

HP-32S





# Help Us Help You!

Please take a moment to complete this postage-paid card, tear it out and put it in the mail. Your responses and comments will help us better understand your needs and will provide you with the best procedures to solve your problems. Thank you!

## HELP US HELP YOU!

Book: **Engineering Applications** Date acquired: \_\_\_\_\_

Name \_\_\_\_\_

Street \_\_\_\_\_

City, State, Zip \_\_\_\_\_

Phone (\_\_\_\_\_) \_\_\_\_\_ Business \_\_\_\_\_ or Home \_\_\_\_\_


1. What calculator will you use this book with?  
008  **HP-32S** 006  **Other** \_\_\_\_\_
2. How many other HP solution books have you bought for this calculator? \_\_\_\_\_
3. What is your OCCUPATION?  
101  **Student** 103  **Professional** 109  **Other** \_\_\_\_\_
4. Where did you purchase this book?  
403  **Bookstore** 404  **Discount or Catalog Store**  
407  **Mail Order** 410  **HP Direct** 411  **Other** \_\_\_\_\_
5. How did you first hear about this book?  
501  **HP Owner** 503  **Advertising** 506  **Salesperson** 507  **Brochure**  
508  **Other** \_\_\_\_\_
6. To what degree did this book influence your calculator purchase decision?  
601  **Major Influence** 602  **Minor Influence** 603  **No Influence**
7. How well does this book cover the material you expected?  
701  **Good** 702  **Moderate** 703  **Low**
8. What level of knowledge is required to make use of the topics in this book?  
801  **High** 802  **Medium** 803  **Low**
9. How clearly was the material in this book presented?  
901  **Good** 902  **Moderate** 903  **Low**
10. How would you rate the value of this book for your money?  
111  **High** 112  **Medium** 113  **Low**

**Comments:** (Please comment on improvements and additional applications or subjects you would like HP to cover in this or another solution book.) \_\_\_\_\_

---

---

---





NO POSTAGE  
NECESSARY  
IF MAILED  
IN THE  
UNITED STATES

**BUSINESS REPLY MAIL**

FIRST CLASS MAIL PERMIT NO. 40 CORVALLIS, OR

POSTAGE WILL BE PAID BY ADDRESSEE

**HEWLETT-PACKARD COMPANY  
CALCULATOR MARKET RESEARCH  
1000 NE CIRCLE BLVD.  
CORVALLIS OR 97330-9988**



# Engineering Applications

---

## Step-by-Step Solutions for Your HP-32S Calculator



Edition 1 June 1988  
Reorder Number 00032-90057

---

## Notice

This manual and any keystroke programs contained herein are provided "as is" and are subject to change without notice. **Hewlett-Packard Company makes no warranty of any kind with regard to this manual or the keystroke programs contained herein, including, but not limited to, the implied warranties of merchantability and fitness for a particular purpose.** Hewlett-Packard Co. shall not be liable for any errors or for incidental or consequential damages in connection with the furnishing, performance, or use of this manual or the keystroke programs contained herein.

© Hewlett-Packard Co. 1988. All rights reserved. Reproduction, adaptation, or translation of this manual, including any programs, is prohibited without prior written permission of Hewlett-Packard Company, except as allowed under the copyright laws. Hewlett-Packard Company grants you the right to use any program contained in this manual in this Hewlett-Packard calculator.

The programs that control your calculator are copyrighted and all rights are reserved. Reproduction, adaptation, or translation of those programs without prior written permission of Hewlett-Packard Co. is also prohibited.

**Corvallis Division  
1000 N.E. Circle Blvd.  
Corvallis, OR 97330, U.S.A.**

---

## Printing History

**Edition 1**

**June 1988**

**Mfg. No. 00032-90058**

# Contents

---

**5    How to Use This Book**

---

<b>1</b>	<b>7</b>	<b>Electrical Engineering</b>
	<b>8</b>	Reactance Chart
	<b>11</b>	Impedance of a Ladder Network
	<b>15</b>	Smith Chart Conversions
	<b>21</b>	Transistor Amplifier Performance

---

<b>2</b>	<b>26</b>	<b>Mechanical Engineering</b>
	<b>27</b>	Black Body Thermal Radiation
	<b>32</b>	Ideal Gas Equation
	<b>35</b>	Conduit Flow
	<b>40</b>	Static Equivalent at a Point
	<b>45</b>	Composite Section Properties
	<b>52</b>	Soderberg's Equation for Fatigue

---

<b>3</b>	<b>55</b>	<b>Civil Engineering</b>
	<b>56</b>	Mohr's Circle for Stress
	<b>63</b>	Field Angle Traverse

---

<b>4</b>	<b>69</b>	<b>Statistics</b>
	<b>70</b>	t Statistics
	<b>70</b>	Paired t Statistics
	<b>71</b>	t Statistic for Two Means
	<b>76</b>	Chi-Square Evaluation
	<b>80</b>	F Distribution
	<b>84</b>	Analysis of Variance (One Way)
	<b>88</b>	Binomial Distribution
	<b>92</b>	Poisson Distribution

---

<b>5</b>	<b>95</b>	<b>Mathematics</b>
	<b>96</b>	Triangle Solutions
	<b>102</b>	Derivative of a Function
	<b>104</b>	Linear Interpolation
	<b>106</b>	Circle Determined by Three Points



# How to Use This Book

---

The *Engineering Applications* solutions book provides sets of keystrokes and routines to help you solve a variety of engineering, statistics, and mathematics problems. The routines have been written to provide for easy use and minimum memory space. This book is to be used with the HP-32S calculator.

Before you use the solutions in this book, you should be familiar with the following concepts from the owner's manual:

- The basics of your calculator — how to perform arithmetic operations, move from menu to menu, and use the menu keys to do calculations.
- How to use the SOLVE function to solve for a variable.
- How to enter numbers for statistics.
- How to key in and run a program. You may wish to refer to the Function Index in your *HP-32S Owner's Manual* for information on how to key a particular function into a program.
- How to determine the number of bytes in a program and how to display the checksum.

**Keys and Menu Selection.** A key on the calculator keyboard is represented like this: **[STO]** . A shifted function is preceded by a shift key, like this: **■ [ASIN]** . A menu label is represented like this: **{DSE}** . It is often necessary to go through several menus to obtain the desired function. For example: **■ [TESTS] {x?y} {>y}** .

**Display Formats.** The examples in this book show numbers displayed to four decimal places. You may change the number of decimal places your calculator displays by pressing **■** **[DISP]** **{F $\times$ }** and the number of decimal places desired. If you wish to see the full 12-digit precision of a number regardless of the display format, press **■** **[SHOW]**; the full precision number is displayed as long as you hold down the **[SHOW]** key.

**Programs.** The HP-32S calculator uses single letters to denote program labels; you have up to 26 labels in program memory. When keying in the program listings in this book, your calculator will display a **DUPLICAT. LBL** error if you use a letter for a label that is already used in program memory. To avoid this problem, simply choose a different letter to designate the label. Be sure to change any **[XEQ]** or **[GTO]** statements that correspond to the newly-assigned label and make note of the changes so that when you execute the routine, you specify the proper label.

When you key in a number having more than three digits in its mantissa, the HP-32S automatically inserts appropriate commas into the number in both data-entry and programming modes.



**Note**

Changing the label name of a program affects its checksum.

---

**Checksum.** A checksum is provided for each program listing as a verification that the program has been keyed in correctly. To view the checksum, press **■** **[MEM]** **{PGM}** and scroll through the listing to the program label you want to check. Press and hold **■** **[SHOW]** to display the checksum.

*Our thanks to Tony Vogt of Oregon State University for developing the problems and equations in this book.*

# **Electrical Engineering**

---

---

## Reactance Chart

This program calculates the resonant frequency, the inductance, or the capacitance of an LC circuit at resonance given the other two variables. It also calculates the capacitive and inductive reactances at resonance, which are equal.

$$f = \frac{1}{2\pi\sqrt{LC}}$$

$$X = \frac{1}{2\pi fC}$$

where:

$L$  = inductance in henrys.

$C$  = capacitance in microfarads.

$f$  = resonant frequency in hertz.

$X$  = reactance in ohms.

## Program Listing.

When keying in steps R12 and R13, press 6 **ENTER** **←** and **+/-**.  
These steps require less memory than keying in -6.

R01 LBL R	R12 6
R02 INPUT F	R13 +/-
R03 INPUT L	R14 10 <sup>∞</sup>
R04 INPUT C	R15 ×
R05 2	R16 RCL× C
R06 π	R17 RCL× W
R07 ×	R18 1
R08 RCL× F	R19 -
R09 STO W	R20 RTN
R10 RCL× L	Checksum = 6CD2
R11 STO X	

**Flags Used.** None.

**Memory Required.** 30 bytes.

**Remarks.** The value of  $C$  is in microfarads to increase the precision of the SOLVE function.

### Program Instructions.

1. Key in program listing; press **C** when finished.
2. Press **■ SOLVE/∫** {FN} R.
3. Specify the unknown variable by pressing **■ SOLVE/∫** {SOLVE} *variable*.
4. Key in the variable value at each prompt and press **R/S**.
5. See the variable for which the program is solving.
6. Press **■ VIEW** X to see the reactance.
7. For a new case, go to step 3.

## Variables Used.

$L$  = inductance in henrys.

$C$  = capacitance in microfarads.

$F$  = resonant frequency in hertz.

$X$  = reactance in ohms.

$W = 2\pi f$  (angular velocity  $\omega$  in radians per second).

## Example. Resonant Frequency and Reactance.

Calculate  $F$  and  $X$ , when  $L = 1.0$  mh and  $C = 0.25$   $\mu$ f.

Keys:	Display:	Description:
■ <b>SOLVE/f</b> {FN}	FN = _	Prompts for program label.
R	<i>value</i>	Specifies program R.
■ <b>SOLVE/f</b> {SOLVE}	SOLVE _	Prompts for the unknown variable.
F	L? <i>value</i>	Starts program R; prompts for variables <i>except F</i> .
<b>E</b> 3 <b>+/-</b> <b>R/S</b>	C? <i>value</i>	$C$ must be in microfarads.
.25 <b>R/S</b>	F = 10,065.8424	Displays the resonant frequency.
■ <b>VIEW</b> X	X = 63.2456	Displays the reactance.

---

## Impedance of a Ladder Network

This program computes the input impedance of a ladder network. Elements are added one at a time from right to left. The first element must be parallel. The input impedance may be viewed at any point in the ladder as the elements are added.

Given an input impedance of  $Y_{in}$ , adding a shunt (parallel)  $R$ ,  $L$ , or  $C$  results in a new input impedance of:

$$Y_{new} = \begin{cases} Y_{in} + \left( \frac{1}{R_p} + j0 \right) \\ Y_{in} + \left( 0 - j\frac{1}{\omega L_p} \right) \\ Y_{in} + (0 + j\omega C_p) \end{cases}$$

Adding a series  $R$ ,  $L$ , or  $C$ , we have:

$$Y_{new} = \begin{cases} \left( \left[ \frac{1}{Y_{in}} + (R_s + j0) \right]^{-1} \right) \\ \left( \left[ \frac{1}{Y_{in}} + (0 + j\omega L_s) \right]^{-1} \right) \\ \left( \left[ \frac{1}{Y_{in}} + \left( 0 - j\frac{1}{\omega C_s} \right) \right]^{-1} \right) \end{cases}$$

where  $Z = \frac{1}{Y}$  and  $\omega = 2\pi f$ .

## Program Listing.

N01 LBL N	Checksum = 10CA
N02 INPUT F	L01 LBL L
N03 2	L02 INPUT L
N04 ×	L03 RCL× W
N05 $\pi$	L04 1/×
N06 ×	L05 +/-
N07 STO W	L06 0
N08 0	L07 RTN
N09 ENTER	Checksum = 517E
N10 RTN	R01 LBL R
Checksum = 026C	R02 0
S01 LBL S	R03 INPUT R
S02 CMPLX1/×	R04 1/×
S03 R↓	R05 RTN
S04 R↓	Checksum = F867
S05 CMPLX1/×	Z01 LBL Z
S06 CMPLX+	Z02 CMPLX1/×
S07 CMPLX1/×	Z03 y,x→0,r
S08 RTN	Z04 STO Z
Checksum = 9EA3	Z05 x<>y
P01 LBL P	Z06 STO A
P02 CMPLX+	Z07 VIEW A
P03 RTN	Z08 VIEW Z
Checksum = 9583	Z09 x<>y
C01 LBL C	Z10 0,r→y,x
C02 INPUT C	Z11 CMPLX1/×
C03 RCL× W	Z12 RTN
C04 0	Checksum = 5450
C05 RTN	

**Flags Used.** None.

**Memory Required.** 75 bytes.



## Remarks.

- The program performs calculations using the admittance in *cartesian* coordinates but displays the result as an impedance in *polar* coordinates.
- Angles must be consistent with the angular mode currently set in the calculator.

## Program Instructions.

1. Key in the program listing and press [C] when finished.
2. Press [XEQ] N.
3. Key in the frequency and press [R/S] .
4. Select the appropriate element to add:
  - Press [XEQ] R to add a resistor.
  - Press [XEQ] L to add an inductor.
  - Press [XEQ] C to add a capacitor.
5. Key in the value at the prompt and press [R/S] .
6. Select the appropriate means of adding the element:
  - Press [XEQ] P to add the element in parallel.
  - Press [XEQ] S to add the element in series.
7. To add another element, go to step 4.
8. Press [XEQ] Z to see the angle of the input impedance.
9. Press [R/S] to see the magnitude of the input impedance.
10. Optional: press [R/S] to continue adding elements to the ladder.

## Variables Used.

$R$  = resistance in ohms.

$L$  = inductance in henrys.

$C$  = capacitance in farads.


$Z$  = magnitude of the input impedance in ohms.

$A$  = input impedance angle.

$F$  = frequency in hertz.

$W = 2\pi f$  (angular velocity  $\omega$  in radians per second).

**Example: RLC Ladder Network.** Find the input impedance of the following circuit at a frequency of 1 MHz:



Keys:	Display:	Description:
<b>MODES</b> {D/G}		Sets <i>degrees</i> mode.
<b>XEQ</b> N	F?value	Inputs frequency.
<b>E</b> 6 <b>R/S</b>	0.0000	
<b>XEQ</b> R	R?value	Adds resistor in parallel
100 <b>R/S</b>	0.0100	(first element must be in parallel).
<b>XEQ</b> P	0.0100	
<b>XEQ</b> C	C?value	Adds capacitor in series.
650 <b>E</b> 12 <b>+/-</b> <b>R/S</b>	0.0000	
<b>XEQ</b> S	0.0014	
<b>XEQ</b> L	L?value	Adds inductor in parallel.
120 <b>E</b> 6 <b>+/-</b> <b>R/S</b>	0.0000	
<b>XEQ</b> P	0.0014	
<b>XEQ</b> R	R? 100.0000	Adds resistor in parallel.
1000 <b>R/S</b>	0.0010	
<b>XEQ</b> P	0.0024	
<b>XEQ</b> Z	A = -41.8224	Displays the input impedance angle.
<b>R/S</b>	Z = 306.7333	Displays the input impedance.

# Smith Chart Conversions

The distance between a point on a Smith Chart and its center may be measured using a number of parameters. This program performs conversions between several of the most commonly used parameters: standing wave ratio, reflection coefficient, and return loss. It may also be used to convert between impedance and reflection coefficient.

$$\sigma = \text{voltage standing wave ratio} = \frac{1 + \rho}{1 - \rho}$$

*SWR* = standing wave ratio expressed in decibels.

$\rho$  = reflection coefficient.


*R.L.* = return loss.

These parameters are related as follows:


$$SWR = 20 \log \sigma$$

$$R.L. = 20 \log \frac{1}{\rho}$$

$$\sigma = \frac{1 + \rho}{1 - \rho}$$



These relationships are perhaps more clearly seen in this sketch:



For a system having characteristic impedance  $Z_0$ , the impedance and reflection coefficient are related by

$$\Gamma = \rho \angle \phi = \frac{\frac{Z}{Z_0} - 1}{\frac{Z}{Z_0} + 1}$$

and

$$Z = Z_0 \angle \theta = Z_0 \frac{1 + \Gamma}{1 - \Gamma}$$

where:

$\Gamma$  = complex reflection coefficient.

$\rho = |\Gamma|$ .

$\phi = \angle \Gamma$ .

$Z$  = impedance.

$Z = |Z|$ .

$\theta = \angle Z$ .

## Program Listing.

A01 LBL A	H02 1
A02 +/-	H03 STO K
Checksum = 13CE	H04 INPUT M
D01 LBL D	H05 INPUT P
D02 20	H06 INPUT D
D03 ÷	H07 INPUT C
D04 10 <sup>x</sup>	H08 RCL M
D05 RTN	H09 RCL P
Checksum = B0E3	H10 $\theta, r \rightarrow y, x$
B01 LBL B	H11 RCL+ K
B02 STO K	H12 RCL M
B03 1	H13 RCL P
B04 +	H14 +/-
B05 1	H15 $\theta, r \rightarrow y, x$
B06 RCL- K	H16 RCL+ K
B07 ÷	H17 CMLPX÷
B08 RTN	H18 RCL D
Checksum = 68C2	H19 RCL C
E01 LBL E	H20 $\theta, r \rightarrow y, x$
E02 STO K	H21 CMLPX $\times$
E03 1	H22 $y, x \rightarrow \theta, r$
E04 -	H23 STO Z
E05 1	H24 $x \langle \rangle y$
E06 RCL+ K	H25 STO T
E07 ÷	H26 VIEW T
E08 RTN	H27 VIEW Z
Checksum = 17DB	H28 RTN
F01 LBL F	Checksum = E61D
F02 1/ $x$	G01 LBL G
Checksum = 7789	G02 INPUT T
C01 LBL C	G03 INPUT Z
C02 LOG	G04 INPUT D
C03 20	G05 INPUT C
C04 $\times$	G06 RCL T
C05 RTN	G07 RCL Z
Checksum = 5C10	G08 $\theta, r \rightarrow y, x$
H01 LBL H	G09 RCL D

G10 RCL C	G21 RCL+ A
G11 $\theta, r \rightarrow y, x$	G22 CMLX $\div$
G12 CMLX $\div$	G23 $y, x \rightarrow \theta, r$
G13 STO A	G24 STO P
G14 $x \langle \rangle y$	G25 $x \langle \rangle y$
G15 STO B	G26 STO M
G16 $x \langle \rangle y$	G27 VIEW M
G17 1	G28 VIEW P
G18 -	G29 RTN
G19 RCL B	Checksum = 6425
G20 1	

**Flags Used.** None.

**Memory Required.** 130.5 bytes.

**Remarks.**

- Each routine is independent of the others. Therefore, key in only those routines that will be used.
- Angles must be consistent with the angular mode currently set in the calculator.

**Program Instructions.**

1. Key in the program listings of the routines to be used; press  $\boxed{C}$  when finished.
2. For conversions involving real number parameters (routines *A* thru *F*):
  - Key in the variable and select the appropriate routine:
    - Press  $\boxed{XEQ}$  A to convert *R.L.* to  $\rho$ .
    - Press  $\boxed{XEQ}$  B to convert  $\rho$  to  $\sigma$ .
    - Press  $\boxed{XEQ}$  C to convert  $\sigma$  to *SWR*.
    - Press  $\boxed{XEQ}$  D to convert *SWR* to  $\sigma$ .
    - Press  $\boxed{XEQ}$  E to convert  $\sigma$  to  $\rho$ .
    - Press  $\boxed{XEQ}$  F to convert  $\rho$  to *R.L.*
  - Optional: continue converting by executing the next routine in the sequence.

3. To convert from  $Z$  to  $\Gamma$ :

- Press  $\boxed{\text{XEQ}}$  G.
- Key in the values at each prompt and press  $\boxed{\text{R/S}}$  .
- See  $M$ ; press  $\boxed{\text{R/S}}$  ; see  $P$ .

4. To convert from  $\Gamma$  to  $Z$ :

- Press  $\boxed{\text{XEQ}}$  H.
- Key in the variables at each prompt and press  $\boxed{\text{R/S}}$  .
- See  $T$ ; press  $\boxed{\text{R/S}}$  ; see  $Z$ .

5. For a new case, go to step 2, 3, or 4.

### Variables Used.

$Z$  = magnitude of the impedance.

$T$  = angle of the impedance.

$P$  = magnitude of the complex reflection coefficient.

$M$  = angle of the complex reflection coefficient.

$C$  = magnitude of the characteristic impedance.

$D$  = angle of the characteristic impedance.

$A, K, B$  = variables used for intermediate results.

**Example 1.** Convert a 10 dB  $SWR$  to  $\sigma$ .

**Keys:**

10  $\boxed{\text{XEQ}}$  D

**Display:**

3.1623

**Description:**

Displays  $\sigma$ .

**Example 2.** Convert a 5 dB return loss to  $SWR$ .

Keys:	Display:	Description:
5 [XEQ] A	0.5623	Displays $\rho$ .
[XEQ] B	3.5698	Displays $\sigma$ .
[XEQ] C	11.0528	Displays $SWR$ .

**Example 3.** A  $75 \Omega$  system is terminated with an impedance of 53 at an angle of  $41^\circ$ . Find the reflection coefficient.

Keys:	Display:	Description:
■ [MODES] {DG}		Sets <i>degrees</i> mode.
[XEQ] G	T?value	Inputs values.
41 [R/S]	Z?value	
53 [R/S]	D?value	
0 [R/S]	C?value	
75 [R/S]	M = 118.3651	Displays $\Gamma$ .
[R/S]	P = 0.4107	Displays $\rho$ .

**Example 4.** A reflection coefficient of 0.35 at an angle of  $11^\circ$  is observed in a  $100 \Omega$  system. Find the impedance.


Keys:	Display:	Description:
[XEQ] H	M?value	Inputs values.
11 [R/S]	P?value	
.35 [R/S]	D?value	
0 [R/S]	C?value	
100 [R/S]	T = 8.6547	Displays angle.
[R/S]	Z = 203.8784	Displays magnitude of the impedance, Z.



---

## Transistor Amplifier Performance

This program calculates several small-signal properties of a transistor amplifier given the h-parameter matrix and the source and load impedances. The properties computed are the current and voltage gains and the input and output impedances.



Definition of h-parameter matrix:

$$\begin{bmatrix} v_1 \\ i_2 \end{bmatrix} = \begin{bmatrix} h_i & h_r \\ h_f & h_o \end{bmatrix} \begin{bmatrix} i_1 \\ v_2 \end{bmatrix}$$

Current gain:

$$A_i = \frac{i_2}{i_1} = \frac{-h_f}{1 + h_o Z_L}$$

Voltage gain:

$$A_v = \frac{v_2}{v_1} = \frac{A_i Z_L}{Z_{in}}$$

Voltage gain with source resistor:

$$A_{vs} = \frac{v_2}{v_s} = \frac{A_i Z_L}{Z_{in} + Z_S}$$

Input impedance:

$$Z_{in} = h_i + h_r Z_L A_i$$

Output impedance:

$$Z_{out} = \frac{h_i + Z_S}{h_o h_i + h_o Z_S - h_f h_r}$$

### Program Listing.

T01 LBL T	X21 +/-
T02 CLVARS	X22 0,r→y,×
T03 1.012	X23 CMPLX×
T04 STO i	X24 STO N
Checksum = 7FE4	X25 x⟨⟩y
X01 LBL X	X26 STO M
X02 INPUT(i)	X27 x⟨⟩y
X03 ISG i	X28 XEQ V
X04 GT0 X	X29 RCL C
X05 RCL I	X30 RCL+ K
X06 RCL J	X31 RCL D
X07 0,r→y,×	X32 RCL× L
X08 STO J	X33 0,r→y,×
X09 x⟨⟩y	X34 CMPLX×
X10 STO I	X35 RCL A
X11 RCL G	X36 RCL B
X12 RCL+ K	X37 0,r→y,×
X13 RCL H	X38 CMPLX+
X14 RCL× L	X39 STO P
X15 0,r→y,×	X40 x⟨⟩y
X16 1	X41 STO 0
X17 +	X42 x⟨⟩y
X18 CMPLX1/×	X43 CMPLX1/×
X19 RCL E	X44 RCL M
X20 RCL F	X45 RCL N

X46	CMPLX×	X74	$\theta, r \rightarrow y, x$
X47	RCL K	X75	CMPLX+
X48	RCL L	X76	RCL E
X49	$\theta, r \rightarrow y, x$	X77	RCL+ C
X50	CMPLX×	X78	RCL F
X51	XEQ V	X79	RCL× D
X52	RCL O	X80	$\theta, r \rightarrow y, x$
X53	RCL P	X81	CMPLX-
X54	CMPLX×	X82	CMPLX1/×
X55	RCL O	X83	RCL A
X56	RCL+ I	X84	RCL B
X57	RCL P	X85	$\theta, r \rightarrow y, x$
X58	RCL+ J	X86	RCL+ J
X59	CMPLX÷	X87	$x \langle \rangle y$
X60	XEQ V	X88	RCL+ I
X61	RCL O	X89	$x \langle \rangle y$
X62	RCL P	X90	CMPLX×
X63	XEQ V	Checksum = 72DD	
X64	RCL G	V01	LBL V
X65	RCL H	V02	$y, x \rightarrow \theta, r$
X66	$\theta, r \rightarrow y, x$	V03	STO R
X67	RCL I	V04	$x \langle \rangle y$
X68	RCL J	V05	STO T
X69	CMPLX×	V06	$x \langle \rangle y$
X70	RCL G	V07	$\theta, r \rightarrow y, x$
X71	RCL+ A	V08	VIEW T
X72	RCL H	V09	VIEW R
X73	RCL× B	V10	RTN
		Checksum = EFAE	

**Flags Used.** None.

**Memory Required.** 164 bytes.

**Remarks.**

- *This program clears all variables stored in Continuous Memory.*
- Angles must be consistent with the angular mode currently set in the calculator.
- To limit the number of variables, the program uses variable *T* for the angle and *R* for the magnitude of all of the output results.

## Program Instructions.

1. Key in the program listings; press  $\boxed{C}$  when finished.
2. Press  $\boxed{XEQ}$  T.
3. Key in the variables at each prompt and press  $\boxed{R/S}$  .
4. See the angle of  $A_i$  and press  $\boxed{R/S}$  .
5. See the magnitude of  $A_i$  and press  $\boxed{R/S}$  .
6. See the angle of  $A_v$  and press  $\boxed{R/S}$  .
7. See the magnitude of  $A_v$  and press  $\boxed{R/S}$  .
8. See the angle of  $A_{vs}$  and press  $\boxed{R/S}$  .
9. See the magnitude of  $A_{vs}$  and press  $\boxed{R/S}$  .
10. See the angle of  $Z_{in}$  and press  $\boxed{R/S}$  .
11. See the magnitude of  $Z_{in}$  and press  $\boxed{R/S}$  .
12. See the angle of  $Z_{out}$  and press  $\boxed{R/S}$  .
13. See the magnitude of  $Z_{out}$  .
14. For a new case, go to step 2.

## Variables Used.

$A$  = angle of  $h_i$ .

$B$  = magnitude of  $h_i$ .

$C$  = angle of  $h_r$ .

$D$  = magnitude of  $h_r$ .

$E$  = angle of  $h_f$ .

$F$  = magnitude of  $h_f$ .

$G$  = angle of  $h_o$ .

$H$  = magnitude of  $h_o$ .

$I$  = angle of  $Z_{in}$ .

$J$  = magnitude of  $Z_{in}$ .

$K$  = angle of  $Z_{out}$ .

$L$  = magnitude of  $Z_{out}$ .

$T$  = angle of  $A_i, A_v, A_{vs}, Z_{in}, Z_{out}$ .

$R$  = magnitude of  $A_i, A_v, A_{vs}, Z_{in}, Z_{out}$ .

$N, M, O, P, i$  = variables used for intermediate results.

**Example.** Find the small-signal properties of a transistor that has the following h-parameter matrix with source and load impedances of 1000 and 5000 ohms, respectively.

$$h = \begin{bmatrix} 1000 & 150E-6 \\ 75 & 50E-6 \end{bmatrix}$$

<b>Keys:</b>	<b>Display:</b>	<b>Description:</b>
<b>MODES</b> {DG}		Sets <i>degrees</i> mode.
<b>XEQ</b> T	A?0.0000	
<b>R/S</b>	B? 0.0000	Inputs values.
1000 <b>R/S</b>	C? 0.0000	
<b>R/S</b>	D? 0.0000	
150 <b>E</b> 6 <b>+/-</b> <b>R/S</b>	E? 0.0000	
<b>R/S</b>	F? 0.0000	
75 <b>R/S</b>	G? 0.0000	
<b>R/S</b>	H? 0.0000	
50 <b>E</b> 6 <b>+/-</b> <b>R/S</b>	I? 0.0000	
<b>R/S</b>	J? 0.0000	
1000 <b>R/S</b>	K? 0.0000	
<b>R/S</b>	L? 0.0000	
5000 <b>R/S</b>	T = 180.0000	Displays angle of $A_i$ .
<b>R/S</b>	R = 60.0000	Displays magnitude of $A_i$ .
<b>R/S</b>	T = 180.0000	Displays angle of $A_v$ .
<b>R/S</b>	R = 314.1361	Displays magnitude of $A_v$ .
<b>R/S</b>	T = 180.0000	Displays angle of $A_{vs}$ .
<b>R/S</b>	R = 153.4527	Displays magnitude of $A_{vs}$ .
<b>R/S</b>	T = 0.0000	Displays angle of $Z_{in}$ .
<b>R/S</b>	R = 955.0000	Displays magnitude of $Z_{in}$ .
<b>R/S</b>	T = 0.0000	Displays angle of $Z_{out}$ .
<b>R/S</b>	R = 22,535.2113	Displays magnitude of $Z_{out}$ .

# 2

## **Mechanical Engineering**

---


---

# Black Body Thermal Radiation

All bodies emit thermal radiation according to their temperature. The higher the temperature, the more thermal radiation emitted. A black body is one that emits the maximum possible amount of energy at every wavelength for a specified temperature. The figure below represents the black body thermal emission as a function of wavelength.

This program can be used to calculate:

- The wavelength of maximum emissive power for a given temperature.
- The temperature corresponding to a particular wavelength of maximum emissive power.
- The total emissive power for all wavelengths.
- The emissive power at a particular wavelength and temperature.



$$\lambda_{\max} T = c_3$$

$$E_{b(0-\lambda)} = \sigma T^4$$

$$E_{b\lambda} = \frac{2\pi c_1}{\lambda^5 (e^{c_2/\lambda T} - 1)}$$

where:

$\lambda_{\max}$  = wavelength of maximum emissivity in microns.

T = absolute temperature in °R or K.

$E_{b(0-\lambda)}$  = total emissive power in Btu/hr-ft<sup>2</sup> or watts/cm<sup>2</sup>.

$E_{b\lambda}$  = emissive power at  $\lambda$  in Btu/hr-ft<sup>2</sup>- $\mu\text{m}$  or watts/cm<sup>2</sup>- $\mu\text{m}$ .

$c_1 = 1.8887982 \times 10^7 \text{ Btu} - \mu\text{m}^4/\text{hr} - \text{ft}^2 = 5.9544 \times 10^3 \text{ W}\mu\text{m}^4/\text{cm}^2$ .

$c_2 = 2.58984 \times 10^4 \mu\text{m} - ^\circ\text{R} = 1.4388 \times 10^4 \mu\text{m} - \text{K}$ .

$c_3 = 5.216 \times 10^3 \mu\text{m} - ^\circ\text{R} = 2.8978 \times 10^3 \mu\text{m} - \text{K}$ .

$\sigma = 1.713 \times 10^{-9} \text{ Btu/hr} - \text{ft}^2 - ^\circ\text{R}^4 = 5.6693 \times 10^{-12} \text{ W/cm}^2 - \text{K}^4$ .

$\sigma_{\text{exp}} = 1.731 \times 10^{-9} \text{ Btu/hr} - \text{ft}^2 - ^\circ\text{R}^4 = 5.729 \times 10^{-12} \text{ W/cm}^2 - \text{K}^4$ .

## Program Listing.

B01 LBL B	B16 2
B02 INPUT W	B17 ×
B03 INPUT T	B18 π
B04 RCL W	B19 ×
B05 1/×	B20 RCL× A
B06 RCL× B	B21 STO P
B07 RCL÷ T	B22 VIEW P
B08 e <sup>x</sup>	B23 RTN
B09 1	Checksum = 52E5
B10 -	E01 LBL E
B11 1/×	E02 INPUT T
B12 RCL W	E03 4
B13 5	E04 y <sup>x</sup>
B14 y <sup>x</sup>	E05 RCL× S
B15 ÷	E06 STO E



```
E07 VIEW E
E08 RTN
Checksum = 15C6
W01 LBL W
W02 INPUT T
W03 RCL C
W04 RCL ÷ T
W05 STO W
W06 VIEW W
```

```
W07 RTN
Checksum = F2C4
T01 LBL T
T02 INPUT W
T03 RCL C
T04 RCL ÷ W
T05 STO T
T06 VIEW T
T07 RTN
Checksum = CA08
```

**Flags Used.** None.

**Memory Required.** 67.5 bytes.

**Remarks.** The values of the constants differ between sources.

### Program Instructions.

1. Key in the program listing; press **[C]** when finished.
2. Store the constants  $A$ ,  $B$ ,  $C$ , and  $S$  in the appropriate storage registers.
3. Select the appropriate routine:
  - Press **[XEQ]** W to calculate the wavelength of maximum power for a given temperature.
  - Press **[XEQ]** T to calculate the temperature corresponding to a particular wavelength of maximum power.
  - Press **[XEQ]** E to calculate the total emissive power.
  - Press **[XEQ]** B to calculate the emissive power at a particular wavelength.
  - Key in the variables at each prompt and press **[R/S]** .
  - See the variable for which the program is solving.
4. For a new case, go to step 3.

**Variables Used.**

$T$  = temperature.

$W$  = wavelength.

$E$  = total emissive power.

$P$  = emissive power at a given wavelength.

$A$  = constant  $c_1$ .

$B$  = constant  $c_2$ .

$C$  = constant  $c_3$ .

$S$  = constant  $\sigma$ .

**Example.** If sunlight has a maximum wavelength of  $.550 \mu\text{m}$ , what is the sun's temperature in K? Assume the sun is a black body. What is the total emissive power and the emissive power at  $\lambda_{\text{max}}$ ? What is the emissive power at  $\lambda = 0.400 \mu\text{m}$  (ultraviolet limit) and  $0.700 \mu\text{m}$  (infrared limit)?

Keys:	Display:	Description:
5.9544 [E] 3 [STO] A		Stores constants.
1.4388 [E] 4 [STO] B		
2.8978 [E] 3 [STO] C		
5.6693 [E] 12 [+/-]		
[STO] S		
[XEQ] T	$W?value$	
.55 [R/S]	$T=5,268.7273$	Displays temperature.
[XEQ] E	$T?5,268.7273$	
[R/S]	$E=4,368.7009$	Displays the total emissive power.
[XEQ] B	$W?0.5500$	Correct value already stored in $W$ .

**R/S**

T?5,268.7273

**R/S**

P=5,222.8745

Displays the emissive power.

**XEQ** B

W?0.5500

.4 **R/S**

T?5,268.7273

**R/S**

P=3,964.8581

Displays the emissive power.

**XEQ** B

W?0.4000

.7 **R/S**

T?5,268.7273

**R/S**

P=4,593.4033

Displays the emissive power.

---

## Ideal Gas Equation

Many gases obey the ideal gas law at high temperatures and low pressures. This program calculates any one of the four variables of the ideal gas equation when the other three are known.

$$PV = nRT$$

where:

$P$  = pressure.

$V$  = volume.

$n$  = number of moles.

$R$  = Universal Gas Constant.

$T$  = absolute temperature.

**Table 2-1. Values of the Universal Gas Constant**

<b>Value of <math>R</math></b>	<b>Units of <math>R</math></b>	<b>Units of <math>P</math></b>	<b>Units of <math>V</math></b>	<b>Units of <math>T</math></b>
8.314	N-m/g mole-K	N/M <sup>2</sup>	m <sup>3</sup> /g mole	K
83.14	cm <sup>3</sup> -bar/g mole-K	bar	cm <sup>3</sup> /g mole	K
82.05	cm <sup>3</sup> -atm/g mole-K	atm	cm <sup>3</sup> /g mole	K
0.08205	liter-atm/g mole-K	atm	liter/g mole	K
0.7302	atm-ft <sup>3</sup> /lbm mole-°R	atm	ft <sup>3</sup> /lbm mole	°R
10.73	psi-ft <sup>3</sup> /lbm mole-°R	psi	ft <sup>3</sup> /lbm mole	°R
1545	psf-ft <sup>3</sup> /lbm mole-°R	psf	ft <sup>3</sup> /lbm mole	°R

## Program Listing.

G01 LBL G	G08 RCL× V
G02 INPUT P	G09 RCL N
G03 INPUT V	G10 RCL× R
G04 INPUT N	G11 RCL× T
G05 INPUT R	G12 -
G06 INPUT T	G13 RTN
G07 RCL P	Checksum = 63C5

**Flags Used.** None.

**Memory Required.** 19.5 bytes.

**Remarks.** Value of  $R$  must be compatible with units of  $P$ ,  $V$ , and  $T$ .

## Program Instructions.

1. Key in program listing; press **[C]** when finished.
2. Press **[SOLVE/F]** {FH} G, then specify the unknown variable by pressing **[SOLVE/F]** {SOLVE} *variable*.
3. Key in the variables at each prompt and press **[R/S]** .
4. See the variable for which the program is solving.
5. For a new case, go to step 2.

## Variables Used.

$P$  = absolute pressure.

$V$  = volume.

$N$  = number of moles present.

$R$  = Universal Gas Constant.

$T$  = absolute temperature.

**Example 1: Pressure.** If 1.2 moles of air are enclosed in 40,000 cm<sup>3</sup> at 1500 K, what is the pressure in atmospheres?

<b>Keys:</b>	<b>Display:</b>	<b>Description:</b>
■ <span style="border: 1px solid black; padding: 2px;">SOLVE/F</span> {FN}	FN= _	Prompts for program label.
G		Specifies program G.
■ <span style="border: 1px solid black; padding: 2px;">SOLVE/F</span> {SOLVE}	SOLVE _	Prompts for the unknown variable.
P	V?value	Starts program G; prompts for variables <i>except P</i> .
40000 <span style="border: 1px solid black; padding: 2px;">R/S</span>	N?value	
1.2 <span style="border: 1px solid black; padding: 2px;">R/S</span>	R?value	
82.05 <span style="border: 1px solid black; padding: 2px;">R/S</span>	T?value	
1500 <span style="border: 1px solid black; padding: 2px;">R/S</span>	P=3.6923	

**Example 2: Specific Volume.** What is the specific volume (ft<sup>3</sup>/lbm) of a gas at a pressure of 3 atmospheres and a temperature of 540°R? The molecular weight is 32 lbm/lbm-mole.

<b>Keys:</b>	<b>Display:</b>	<b>Description:</b>
■ <span style="border: 1px solid black; padding: 2px;">SOLVE/F</span> {SOLVE}	SOLVE _	Prompts for the unknown variable. (It is not necessary to redefine the program label being executed since it was defined in the last example.
V	P?value	Starts program G; prompts for the variables <i>except V</i> .
3 <span style="border: 1px solid black; padding: 2px;">R/S</span>	N?value	
32 <span style="border: 1px solid black; padding: 2px;">1/x</span> <span style="border: 1px solid black; padding: 2px;">R/S</span>	R?value	
.7302 <span style="border: 1px solid black; padding: 2px;">R/S</span>	T?value	
540 <span style="border: 1px solid black; padding: 2px;">R/S</span>	V=4.1074	

---

## Conduit Flow

This program solves for either the average velocity or the pressure drop for viscous, incompressible flow in conduits.

For laminar flow ( $Re < 2300$ ):

$$f = 16/Re$$

For turbulent flow ( $Re > 2300$ ):

$$V^2 = \frac{\Delta P / \rho}{2 \left( f \frac{L}{D} + \frac{K_T}{4} \right)}$$
$$f = \frac{0.0772}{\left\{ \log \left[ \frac{6.9}{Re} + \left( \frac{\epsilon}{3.7D} \right)^{1.111} \right] \right\}^2}$$

where:

$V$  = average velocity.

$\Delta P$  = pressure drop.

$L$  = conduit length.

$D$  = conduit diameter. If the conduit is *not* circular, use an *equivalent* diameter defined by:

$$D_{eq} = 4 \times \frac{\text{Cross Sectional Area}}{\text{Wetted Perimeter}}$$

$\epsilon$  = surface irregularity.

$Re$  = Reynolds number;  $Re = DV / \nu$ .

$\nu$  = fluid kinematic viscosity.

$\rho$  = fluid density

$f$  = Fanning friction factor.

$K_T$  = sum of fitting factors.

**Table 2-2. Fitting Coefficients**

Fitting	K
Globe valve, wide open	7.5 - 10
Angle valve, wide open	3.8
Gate valve, wide open	0.15 - 0.19
Gate valve, $3/4$ open	0.85
Gate valve, $1/2$ open	4.4
Gate valve, $1/4$ open	20
90° elbow	0.4 - 0.9
Standard 45° elbow	0.35 - 0.42
Tee, through side outlet	1.5
Tee, straight through	0.4
180° bend	1.6
Entrance to circular pipe	0.25 - 0.50
Sudden expansion	$(1 - A_{up}/A_{dn})^2 *$
Acceleration from $V = 0$ to $V = V_{entrance}$	1.0
* $A_{up}$ is the upstream area and $A_{dn}$ is the downstream area.	

**Table 2-3. Surface Irregularities**

Material	$\epsilon$ (Feet)	$\epsilon$ (Meters)
Drawn or smooth tubing	$5.0 \times 10^6$	$1.5 \times 10^6$
Commercial steel or wrought iron	$1.5 \times 10^4$	$4.6 \times 10^5$
Asphalted cast iron	$4.0 \times 10^4$	$1.2 \times 10^4$
Galvanized iron	$5.0 \times 10^4$	$1.5 \times 10^4$
Cast iron	$8.3 \times 10^4$	$2.5 \times 10^4$
Wood stave	$6.0 \times 10^4$ to $3.0 \times 10^3$	$1.8 \times 10^4$ to $9.1 \times 10^4$
Concrete	$1.0 \times 10^3$ to $1.0 \times 10^2$	$3.0 \times 10^4$ to $3.0 \times 10^3$
Riveted steel	$3.0 \times 10^3$ to $3.0 \times 10^2$	$9.1 \times 10^4$ to $9.1 \times 10^3$



## Program Listing.

C01 LBL C	C30 ×
C02 INPUT E	C31 1/x
C03 INPUT D	C32 x <sup>2</sup>
C04 INPUT V	Checksum = 01E3
C05 INPUT B	D01 LBL D
C06 INPUT L	D02 STO F
C07 INPUT K	D03 RCL× L
C08 INPUT S	D04 RCL÷ D
C09 INPUT P	D05 RCL K
C10 RCL E	D06 4
C11 RCL÷ D	D07 ÷
C12 3.7	D08 +
C13 ÷	D09 2
C14 1.111	D10 ×
C15 y <sup>x</sup>	D11 RCL V
C16 RCL V	D12 x <sup>2</sup>
C17 RCL× D	D13 ×
C18 RCL÷ B	D14 RCL P
C19 STO R	D15 RCL÷ S
C20 2,300	D16 -
C21 x>y?	D17 RTN
C22 GT0 L	Checksum = 188C
C23 R+	L01 LBL L
C24 1/x	L02 16
C25 6.9	L03 RCL÷ R
C26 ×	L04 GT0 D
C27 +	L05 RTN
C28 LOG	Checksum = F63D
C29 3.6	

**Flags Used.** None.

**Memory Required.** 121 bytes.

## Program Instructions.

1. Key in program listing; press **[C]** when finished.
2. Press **■ [SOLVE/] {FH} C**, then specify the unknown variable by pressing **■ [SOLVE/] {SOLVE} variable**.
3. Key in the variables at each prompt and press **[R/S]**.
4. See the variable for which the program is solving.
5. Optional: Press **■ [VIEW] R** to see the Reynolds number.
6. Optional: Press **■ [VIEW] F** to see the Fanning friction factor.
7. For a new case, go to step 2.

## Variables Used.

$V$  = average velocity.

$P$  = pressure drop.

$L$  = conduit length.

$D$  = conduit diameter.

$E$  = surface irregularity.

$R$  = Reynolds number.

$B$  = fluid kinematic viscosity.

$S$  = fluid density.

$F$  = Fanning friction factor.

$K$  = fitting coefficient.


**Example: Pressure Drop.** A 60-meter pipe has three 180 degree bends ( $K_T = 3 \times 1.6$ ). The fluid is water ( $\nu = 9.3 \times 10^{-7} \text{ m}^2/\text{s}$ ,  $\rho = 1000 \text{ kg/m}^3$ ). The pipe diameter is 0.030 m and the surface roughness is  $3 \times 10^{-4} \text{ m}$ . If the average velocity is 3.20 m/s, what is the pressure drop in Pascals? What is the Reynolds number? What is the Fanning friction factor?

<b>Keys:</b>	<b>Display:</b>	<b>Description:</b>
<b>■</b> <b>SOLVE</b> / <i>f</i> {FH}	FN= _	Prompts for program label.
C	<i>value</i>	Specifies program C.
<b>■</b> <b>SOLVE</b> / <i>f</i> {SOLVE}	SOLVE _	Prompts for the unknown variable.
P	E? <i>value</i>	Starts program C;
3 <b>E</b> 4 <b>+/-</b> <b>R/S</b>	D? <i>value</i>	prompts for variables
.03 <b>R/S</b>	V? <i>value</i>	<i>except P.</i>
3.2 <b>R/S</b>	B? <i>value</i>	
9.3 <b>E</b> 7 <b>+/-</b> <b>R/S</b>	L? <i>value</i>	
60 <b>R/S</b>	K? <i>value</i>	
4.8 <b>R/S</b>	S? <i>value</i>	
<b>E</b> 3 <b>R/S</b>	P = 418,351.2590	Displays the pressure drop.
<b>■</b> <b>VIEW</b> R	R = 103,225.8065	Displays the Reynolds number.
<b>■</b> <b>VIEW</b> F	F = 0.0096	Displays the friction factor.

---

## Static Equivalent at a Point

This program calculates the two reaction forces necessary to balance any given two-dimensional force vectors, provided the vectors act through the same point. The direction of the reaction forces must be specified as an angle relative to an arbitrary axis.



Equations:

$$R_1 \cos \theta_1 + R_2 \cos \theta_2 = \sum F \cos \phi$$

$$R_1 \sin \theta_1 + R_2 \sin \theta_2 = \sum F \sin \phi$$

where:

$F$  = magnitude of each known force.

$\phi$  = direction of each known force.

$R_1$  = first reaction force.

$\theta_1$  = direction of  $R_1$ .

$R_2$  = second reaction force.

$\theta_2$  = direction of  $R_2$ .

### Program Listing.

```
S01 LBL S
S02 CLVARS
S03 INPUT N
Checksum = CC9D
A01 LBL A
A02 INPUT T
A03 INPUT F
A04 RCL T
A05 RCL F
A06  $\theta, r \rightarrow y, x$ 
A07 STO+ X
A08  $x \langle \rangle y$ 
A09 STO+ Y
A10 DSE N
A11 GTO A
A12 INPUT A
A13 SIN
A14 STO A
A15 LASTx
A16 COS
A17 STO C
A18 INPUT B
A19 SIN
A20 STO B
A21 LASTx
A22 COS
A23 STO D
A24 RCL X
A25 RCLx B
A26 RCL D
A27 RCLx Y
A28 -
A29 RCL A
A30 RCLx D
A31 RCL C
A32 RCLx B
A33 -
A34 ÷
A35 STO R
A36 VIEW R
A37 LASTx
A38 RCL C
```

A39 RCL X Y	A44 ÷
A40 RCL X	A45 STO R
A41 RCL X A	A46 VIEW R
A42 -	A47 RTN
A43 x<>y	Checksum = 665C

**Flags Used.** None.

**Memory Required.** 75 bytes.

**Remarks.**

- *This program clears all variables stored in Continuous Memory.*
- A positive value of force (tension) points away from the origin; a negative value (compression) points toward the origin.
- Angles must be consistent with the angular mode currently set in the calculator.


**Program Instructions.**

1. Key in the program listing; press [C] when finished.
2. Press [XEQ] S.
3. Key in the variables at each prompt and press [R/S] .
4. See the first reaction force, then press [R/S] .
5. See the second reaction force.
6. For a new case, go to step 2.

**Variables Used.**

- $N$  = number of known forces.
- $T$  = angle of each known force.
- $F$  = value of each known force.
- $A$  = direction of the first reaction force.
- $B$  = direction of the second reaction force.
- $R$  = value of the unknown forces  $R_1$  and  $R_2$  .
- $D, X, Y, C$  = variables used for intermediate results.

**Example 1: Balancing a Single Vector.** Find the reaction forces in the following diagram:



**Keys:**

- XEQ S
- 1  R/S
- 135  R/S
- 75  R/S
- 30  R/S
- 270  R/S


**Display:**

- N?0.0000
- T?0.0000
- F?0.0000
- A?0.0000
- B?0.0000
- R=61.2372
- R=83.6516

**Description:**

- Inputs known values.
- Displays the first reaction force.
- Displays the second reaction force.

**Example 2: Forces in a Bridge Truss.** Find the reaction forces in structural members AE and CE. Assume pin connections at the joint.



**Keys:**

XEQ S

2  R/S

45  R/S

100  R/S

180  R/S

120  R/S

R/S

135  R/S

R/S

**Display:**

N?0.0000

T?0.0000

F?0.0000

T?45.0000

F?100.0000

A?0.0000

B?0.0000

R = -21.4214

R = -100.0000

**Description:**

Inputs known values.

Displays the first reaction force.

Displays the second reaction force.



## Composite Section Properties

The mechanical properties of a constant cross section member composed of a finite number of rectangular elements can be computed by adding the contribution of each rectangular region individually. This program uses this principle to calculate the area of a section, the moments of inertia about the specified set of axes, the moments of inertia about an axis translated to the centroid, the moments of inertia of the principal axes, and the angle of rotation between the translated axes and the principal axes.

$$A_{s i} = \Delta x_i \Delta y_i$$

$$A = A_{s 1} + A_{s 2} + A_{s 3} + \cdots + A_{s n}$$

$$\bar{x} = \frac{\sum_{i=1}^n x_{0 i} A_{s i}}{A}$$

$$\bar{y} = \frac{\sum_{i=1}^n y_{0 i} A_{s i}}{A}$$

$$I_{x y} = \sum_{i=1}^n x_{0 i} y_{0 i} A_{s i} \quad I_{\bar{x} \bar{y}} = I_{x y} - A \bar{x} \bar{y}$$

$$I_x = \sum_{i=1}^n \left( y_{0 i}^2 + \frac{\Delta y_i^2}{12} \right) A_{s i} \quad I_{\bar{x}} = I_x - A \bar{y}^2$$

$$I_y = \sum_{i=1}^n \left( x_{0 i}^2 + \frac{\Delta x_i^2}{12} \right) A_{s i} \quad I_{\bar{y}} = I_y - A \bar{x}^2$$

$$J = I_x + I_y$$

$$\phi = \frac{1}{2} \tan^{-1} \left( \frac{2 I_{\bar{x} \bar{y}}}{I_{\bar{x}} - I_{\bar{y}}} \right)$$

$$I_{\bar{x} \phi} = I_{\bar{x}} \cos^2 \phi + I_{\bar{y}} \sin^2 \phi + I_{\bar{x} \bar{y}} \sin 2 \phi$$

$$I_{\bar{y} \phi} = I_{\bar{y}} \cos^2 \phi + I_{\bar{x}} \sin^2 \phi + I_{\bar{x} \bar{y}} \sin 2 \phi$$

$$J_{\phi} = I_{\bar{x} \phi} + I_{\bar{y} \phi}$$

where:

$\Delta x_i$  = width of a rectangular element.

$\Delta y_i$  = height of a rectangular element.

$A_{ei}$  = area of an element.

$A$  = total area of the section.

$\bar{x}$  =  $x$ -coordinate of the centroid.

$\bar{y}$  =  $y$ -coordinate of the centroid.

$x_{\alpha}$  =  $x$ -coordinate of the centroid.

$y_{\alpha}$  =  $y$ -coordinate of the centroid.

$I_x$  = moment of inertia about the  $x$ -axis.

$I_y$  = moment of inertia about the  $y$ -axis.

$J$  = polar moment of inertia about the origin.

$I_{xy}$  = product of inertia about the origin.

$I_{\bar{x}}$  = moment of inertia about the  $x$ -axis translated to the centroid.

$I_{\bar{y}}$  = moment of inertia about the  $y$ -axis translated to the centroid.

$I_{\bar{xy}}$  = product of inertia about the translated axis.

$\phi$  = angle between the translated axis and the principal axis.

$I_{\bar{x}\phi}$  = moment of inertia about the principal  $x$ -axis.

$I_{\bar{y}\phi}$  = moment of inertia about the principal  $y$ -axis.

$J_{\phi}$  = polar moment of inertia about the principal axis.

## Program Listing.

```
S01 LBL S
S02 CLVARS
S03 INPUT N
Checksum = CC9D
U01 LBL U
U02 INPUT X
U03 INPUT Y
U04 INPUT S
U05 INPUT T
U06 RCL× S
U07 STO B
U08 STO+ A
U09 RCL× Y
U10 STO+ D
U11 RCL X
U12 RCL× B
U13 STO+ C
U14 RCL Y
U15 ×²
U16 RCL T
U17 ×²
U18 12
U19 ÷
U20 +
U21 RCL× B
U22 STO+ H
U23 RCL X
U24 ×²
U25 RCL S
U26 ×²
U27 12
U28 ÷
U29 +
U30 RCL× B
U31 STO+ I
U32 +
U33 STO+ J
U34 RCL X
U35 RCL× Y
U36 RCL× B
U37 STO+ P
U38 DSE N
U39 GTO U
U40 RCL P
U41 RCL C
U42 RCL× D
U43 RCL÷ A
U44 -
U45 STO Q
U46 RCL C
U47 RCL÷ A
U48 VIEW A
U49 STO X
U50 VIEW X
U51 ×²
U52 RCL D
U53 RCL÷ A
U54 STO Y
U55 VIEW Y
U56 ×²
U57 RCL× A
U58 +/-
U59 RCL H
U60 VIEW H
U61 +
U62 STO H
U63 R+
U64 RCL× A
U65 +/-
U66 RCL I
U67 VIEW I
U68 VIEW J
U69 VIEW P
U70 +
```

U71 STO I	V01 LBL V
U72 RCL H	V02 STO G
U73 STO J	V03 VIEW G
U74 VIEW H	V04 2
U75 $x \langle \rangle y$	V05 $\times$
U76 VIEW I	V06 SIN
U77 STO+ J	V07 RCL $\times$ P
U78 -	V08 +/-
U79 STO D	V09 RCL G
U80 RCL Q	V10 SIN
U81 STO P	V11 $\times^2$
U82 VIEW P	V12 RCL $\times$ D
U83 $x=0?$	V13 -
U84 GTO V	V14 RCL+ H
U85 $\div$	V15 STO H
U86 $1/\times$	V16 VIEW H
U87 2	V17 RCL- J
U88 $\times$	V18 +/-
U89 ATAN	V19 STO I
U90 2	V20 VIEW I
U91 $\div$	V21 VIEW J
U92 +/-	V22 RTN
Checksum = A4B6	Checksum = 63B7

**Flags Used.** None.

**Memory Required.** 175.5 bytes.

**Remarks.**

- *This program clears all variables stored in Continuous Memory.*
- For a given origin, the polar moment of inertia is constant regardless of the angular rotation. Therefore,  $J_{xy}$  is equal to  $J_{\phi}$ .
- It is possible to obtain a negative value for the product of inertia.

### Program Instructions.

1. Key in the program listing; press **[C]** when done.
2. Press **[XEQ] S**.
3. Key in the variables at each prompt and press **[R/S]**.
4. See the results as they are displayed and press **[R/S]**.
5. For a new case, go to step 2.

### Variables Used.

$$X = x_{\alpha} \text{ and } \bar{x}.$$

$$Y = y_{\alpha} \text{ and } \bar{y}.$$

$$S = \Delta x_i.$$

$$T = \Delta y_i.$$

$$H = I_x, I_{\bar{x}}, \text{ and } I_{\bar{x}\phi}.$$

$$I = I_y, I_{\bar{y}}, \text{ and } I_{\bar{y}\phi}.$$

$$J = J \text{ and } J_{\phi}.$$

$$P = I_{xy}, I_{\bar{x}\bar{y}}.$$


$A$  = total area of the entire section.

$G$  = angle between the translated axis and the principal axis.

$N$  = number of sections.

$D, C, B, Q$  = variables used for intermediate results.

**Example 1: Rectangular Section.** Calculate the section properties of the following cross section:



### Table of Inputs

x	y	$\Delta x$	$\Delta y$
2	1.5	4	3

#### Keys:

**XEQ** S

1 **R/S**

2 **R/S**

1.5 **R/S**

4 **R/S**

3 **R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

#### Display:

N?0.0000

X?0.0000

Y?0.0000

S?0.0000

T?0.0000

A=12.0000

X=2.0000

Y=1.5000

H=36.0000

I=64.0000

J=100.0000

P=36.0000

H=9.0000

I=16.0000

P=0.0000

G=0.0000

H=9.0000

I=16.0000

J=25.0000

#### Description:

Inputs known values.

Displays area.

Displays  $\bar{x}$ .

Displays  $\bar{y}$ .

Displays  $I_x$ .

Displays  $I_y$ .

Displays  $J$ .

Displays  $I_{xy}$ .

Displays  $I_{\bar{x}}$ .

Displays  $I_{\bar{y}}$ .

Displays  $I_{\bar{xy}}$ .


Displays  $\phi$ .

Displays  $I_{\bar{x}\phi}$ .

Displays  $I_{\bar{y}\phi}$ .

Displays  $J_{\phi}$ .

**Example 2: Composite Section.** Calculate the section properties of the following section:



## Table of Inputs

	x	y	$\Delta x$	$\Delta y$
1	5	11.5	6	1
2	1	6	2	12
3	7	1	10	2

### Keys:

### Display:

### Description:

$\boxed{\text{XEQ}}$  S

N?0.0000

3  $\boxed{\text{R/S}}$

X?0.0000

Inputs known values.

5  $\boxed{\text{R/S}}$

Y?0.0000

11.5  $\boxed{\text{R/S}}$

S?0.0000

6  $\boxed{\text{R/S}}$

T?0.0000

1  $\boxed{\text{R/S}}$

X?5.0000

1  $\boxed{\text{R/S}}$

Y?11.5000

6  $\boxed{\text{R/S}}$

S?6.0000

2  $\boxed{\text{R/S}}$

T?1.0000

12  $\boxed{\text{R/S}}$

X?1.0000

7  $\boxed{\text{R/S}}$

Y?6.0000

1  $\boxed{\text{R/S}}$

S?2.0000

10  $\boxed{\text{R/S}}$

T?12.0000

2  $\boxed{\text{R/S}}$

A=50.0000

Displays area.

$\boxed{\text{R/S}}$

X=3.8800

Displays  $\bar{x}$ .

$\boxed{\text{R/S}}$

Y=4.6600

Displays  $\bar{y}$ .

$\boxed{\text{R/S}}$

H=1,972.6667

Displays  $I_x$ .

$\boxed{\text{R/S}}$

I=1,346.6667

Displays  $I_y$ .

$\boxed{\text{R/S}}$

J=3,319.3333

Displays  $J$ .

$\boxed{\text{R/S}}$

P=629.0000

Displays  $I_{xy}$ .

$\boxed{\text{R/S}}$

H=886.8867

Displays  $I_{\bar{x}}$ .

$\boxed{\text{R/S}}$

I=593.9467

Displays  $I_{\bar{y}}$ .

$\boxed{\text{R/S}}$

P=-275.0400

Displays  $I_{\bar{xy}}$ .

$\boxed{\text{R/S}}$

G=30.9814

Displays  $\phi$ .

$\boxed{\text{R/S}}$

H=1,052.0261

Displays  $I_{\bar{x}\phi}$ .

$\boxed{\text{R/S}}$

I=428.8072

Displays  $I_{\bar{x}\phi}$ .


$\boxed{\text{R/S}}$

J=1,480.8333

Displays  $J_{\phi}$ .

# Soderberg's Equation for Fatigue

This program calculates any one of the six variables in Soderberg's equation for fatigue when the other five are known. Soderberg's equation is shown graphically in the figure below.



Equation:

$$\frac{s_{yp}}{FS} = \frac{s_{max} + s_{min}}{2} + K \left( \frac{s_{yp}}{s_e} \right) \left( \frac{s_{max} - s_{min}}{2} \right)$$

where:

$s_{yp}$  = yield point stress.

$s_e$  = endurance stress from reversed bending tests.

$s_{max}$  = maximum applied stress.

$s_{min}$  = minimum applied stress.



$K$  = stress concentration factor.

$FS$  = factor of safety.

### Program Listing.

S01 LBL S	S13 ÷
S02 INPUT Y	S14 -
S03 INPUT E	S15 RCL A
S04 INPUT A	S16 RCL- B
S05 INPUT B	S17 2
S06 INPUT K	S18 ÷
S07 INPUT F	S19 RCL× Y
S08 RCL Y	S20 RCL÷ E
S09 RCL÷ F	S21 RCL× K
S10 RCL A	S22 -
S11 RCL+ B	S23 RTN
S12 2	Checksum = C95D

**Flags Used.** None.

**Memory Required.** 34.5 bytes.

### Remarks.

- Soderberg's equation is valid for ductile materials only.
- Fatigue effects are magnified in corrosive environments.

### Program Instructions.

1. Key in the program listing, pressing **[C]** when finished.
2. Press **[SOLVE/I] {FN} S**, then specify the unknown variable by pressing **[SOLVE/I] {SOLVE} variable**.
3. Key in the variables at each prompt and press **[R/S]**.
4. See the variable for which the program is solving.
5. For a new case, go to step 2.

## Variables Used.

$Y$  = yield point stress.

$E$  = endurance stress.

$A$  = maximum applied stress.

$B$  = minimum applied stress.

$K$  = stress concentration factor.

$F$  = factor of safety.

**Example.** What is the maximum allowable applied stress if the minimum applied stress is 15,000 psi?

$s_{yp} = 80,000$  psi.

$s_e = 30,000$  psi.

$K = 1.5$ .

$FS = 2.0$ .

Keys:	Display:	Description:
■ <b>SOLVE/F</b> {FN}	FN= _	Prompts for program label.
S	<i>value</i>	Specifies program S.
■ <b>SOLVE/F</b> {SOLVE}	SOLVE _	Prompts for the unknown variable.
A	Y? <i>value</i>	Starts program S; prompts for variables <i>except A</i> .
80000 <b>R/S</b>	E? <i>value</i>	
30000 <b>R/S</b>	B? <i>value</i>	
15000 <b>R/S</b>	K? <i>value</i>	
1.5 <b>R/S</b>	F? <i>value</i>	
2 <b>R/S</b>	A=25,000.0000	Displays the maximum applied stress.


# **Civil Engineering**

---

# Mohr's Circle for Stress

This program calculates the 2-D Mohr's circle for stress from equiangular or rectangular strain gage data or directly from known stresses.

Configuration Code	1	2
Type of Rosette	Rectangular	Delta (Equiangular)
Principal Strains $\epsilon_1, \epsilon_2$	$\frac{1}{2} [\epsilon_a + \epsilon_c \pm \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2}]$	$\frac{1}{3} [\epsilon_a + \epsilon_b + \epsilon_c \pm \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2 + 2(\epsilon_c - \epsilon_a)^2}]$
Center of Mohr Circle $\frac{s_1 + s_2}{2}$	$\frac{E(\epsilon_a + \epsilon_c)}{2(1 - \nu)}$	$\frac{E(\epsilon_a + \epsilon_b + \epsilon_c)}{3(1 - \nu)}$
Maximum Shear Stress $\tau_{max}$	$\frac{E}{2(1 + \nu)} \times \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2}$	$\frac{E}{3(1 + \nu)} \times \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2 + 2(\epsilon_c - \epsilon_a)^2}$
Orientation of Principal Stresses $\Delta$	$\frac{1}{2} \tan^{-1} \left[ \frac{2\epsilon_b - \epsilon_a - \epsilon_c}{\epsilon_a - \epsilon_c} \right]$	$\frac{1}{2} \tan^{-1} \left[ \frac{\sqrt{3} (\epsilon_c - \epsilon_b)}{(2\epsilon_a - \epsilon_b - \epsilon_c)} \right]$



$$\tau_{\max} = \left[ \left( \frac{S_x - S_y}{2} \right)^2 + \tau_{xy}^2 \right]^{\frac{1}{2}}$$

$$S_1 = \frac{S_x + S_y}{2} + \tau_{\max}$$

$$S_2 = \frac{S_x + S_y}{2} - \tau_{\max}$$

$$\theta = \frac{1}{2} \tan^{-1} \left( \frac{2\tau_{xy}}{S_x - S_y} \right)$$

$$S = \frac{S_1 + S_2}{2} + \tau_{\max} \cos 2\theta'$$

$$\tau = \tau_{\max} \sin 2\theta'$$

## Program Listing.

```
I01 LBL I
I02 INPUT E
I03 INPUT V
I04 INPUT A
I05 INPUT B
I06 INPUT C
I07 RCL E
I08 1
I09 RCL- V
I10 ÷
I11 STO J
I12 RCL E
I13 1
I14 RCL+ V
I15 ÷
I16 STO R
I17 RCL A
I18 RCL B
I19 RCL C
I20 RTN
Checksum = 9144
R01 LBL R
R02 XEQ I
R03 RCL+ A
R04 2
R05 ÷
R06 RCL- B
R07 RCL C
R08 RCL A
R09 GTO M
Checksum = FFAR
E01 LBL E
E02 XEQ I
E03 -
E04 3
E05 SORT
E06 ÷
E07 RCL B
E08 RCL+ C
E09 2
E10 ×
E11 RCL- A
E12 3
E13 ÷
E14 RCL A
E15 GTO M
Checksum = 8A60
S01 LBL S
S02 INPUT S
S03 INPUT Y
S04 INPUT X
S05 1
S06 STO J
S07 STO R
S08 RCL S
S09 +/-
S10 RCL Y
S11 RCL X
Checksum = F899
M01 LBL M
M02 STO L
M03 +
M04 2
M05 ÷
M06 STO× J
M07 RCL- L
M08 ABS
M09 y,x→θ,r
M10 STO× R
M11 x<>y
M12 2
M13 ÷
M14 STO G
M15 RCL R
```

M16 RCL+ J	M29 SIN
M17 STO U	M30 LAST $\times$
M18 VIEW U	M31 COS
M19 RCL J	M32 RCL $\times$ R
M20 RCL- R	M33 RCL+ J
M21 STO L	M34 STO P
M22 VIEW L	M35 VIEW P
M23 VIEW R	M36 R+
M24 VIEW G	M37 RCL $\times$ R
M25 INPUT W	M38 STO T
M26 RCL+ G	M39 VIEW T
M27 2	M40 RTN
M28 $\times$	Checksum = D351

**Flags Used.** None.

**Memory Required.** 142.5 bytes.

**Remarks.**

- Tensile forces are considered positive, compressive stresses negative.
- This program calculates the principal stresses for a two dimensional stress state only. A knowledge of the stresses in the z-direction is necessary to determine the overall maximum and minimum stresses.
- Angles must be consistent with the angular mode currently set in the calculator.

**Program Instructions.**

1. Key in the program listings of the routines to be used; press **[C]** when finished.
2. Select the appropriate routine:
  - Press **[XEQ] E** if equiangular strain gage readings are known.
  - Press **[XEQ] R** if rectangular strain gage readings are known.
  - Press **[XEQ] S** if stresses are known directly.
3. Key in the variables at each prompt and press **[R/S]** .
4. See each result as it's displayed. Press **[R/S]** to display the next one.
5. Optional: At the prompt, key in rotation angle *W* and press **[R/S]** to obtain the normal stress at that orientation; press **[R/S]** again to see the shear stress.

## Variables Used.

$A = \epsilon_0$ .

$B = \epsilon_{45}$  or  $\epsilon_{60}$ .

$C = \epsilon_{90}$  or  $\epsilon_{120}$ .

$E =$  Young's modulus.

$\nu =$  Poisson's ratio.

$X =$  normal stress on the  $x$ -face,  $\sigma_x$ .

$Y =$  normal stress on the  $y$ -face,  $\sigma_y$ .

$S =$  shear stress,  $\tau_{xy}$ .

$U =$  maximum principal stress,  $\sigma_1$ .

$L =$  minimum principal stress,  $\sigma_2$ .

$R =$  maximum shear stress,  $\tau_{\max}$ .

$G =$  clockwise angle from the specified  $x$ -axis to the maximum principal axis.

$W =$  arbitrary angle counterclockwise from the specified  $x$ -axis,  $\beta$ .

$P =$  normal stress at angle  $\beta$ .

$T =$  shear stress at angle  $\beta$ .

$J =$  variable used for intermediate results.



**Example 1: Equiangular Strain Gage.** An equiangular rosette strain gage measures the following strains:

$$\begin{aligned}\epsilon_0 &= 180 \mu. \\ \epsilon_{60} &= 200 \mu. \\ \epsilon_{120} &= -290 \mu.\end{aligned}$$

Find the principal stresses and their orientation. The material properties are  $E = 30 \times 10^6$  psi and  $\nu = 0.3$ .

**Keys:**

■ **MODES** {D/G}  
**XEQ** E  
 30 **E** 6 **R/S**  
 .3 **R/S**  
 180 **E** 6 **+/-** **R/S**  
 200 **E** 6 **+/-** **R/S**  
 290 **+/-** **E** 6 **+/-**  
**R/S**  
**R/S**  
**R/S**  
**R/S**


**Display:**

E?value  
 V?value  
 A?value  
 B?value  
 C?value  
 U=8,675.1358  
 L= -6,103.7072  
 R=7,389.4215  
 G=31.0333

**Description:**

Sets *degrees* mode.  
 Begins equiangular rosette routine.  
 Inputs strain gage readings.  
 Displays  $\sigma_1$  .  
 Displays  $\sigma_2$  .  
 Displays  $\tau_{max}$  .  
 Displays  $\theta$  .

**Example 2: Known Stresses.** The stresses acting on an element are shown below (all stresses are in MPa).



Find the principal stresses and their orientation, and the stresses on the face of the element oriented  $45^\circ$  counterclockwise from the  $x$ -axis.

**Keys:**

**Display:**

**Description:**

XEQ S

S?value

Begins stress routine.

30  R/S

Y?value

Inputs stresses.

20  +/-  R/S

X?value

75  R/S

U = 83.6805

Displays  $\sigma_1$ .

R/S

L = -28.6805

Displays  $\sigma_2$ .

R/S

R = 56.1805

Displays  $\tau_{max}$ .

R/S

G = -16.1378

Displays  $\theta$ .

R/S

W?value

Inputs angle.

45  R/S

P = 57.5000

Displays  $\sigma$ .

R/S

T = 47.5000

Displays  $\tau$ .

---

## Field Angle Traverse

This program calculates the coordinates of a traverse, the total horizontal distance traversed, and the enclosed area (for a closed traverse). The user must input the northing and easting of the starting point, the reference azimuth, and the direction and distance from each point in the traverse to the next point. The direction may be input either as a deflection right or left or as an angle right or left. The distance may be input either as a horizontal distance or as a slope distance with a zenith angle.

$$HD = SD \sin(ZA)$$

$$N_{k+1} = N_k + HD \cos(AZ)$$

$$E_{k+1} = E_k + HD \sin(AZ)$$

$$LAT_k = N_{k+1} - N_k$$

$$DEP_k = E_{k+1} - E_k$$

$$\text{Area} = \sum_{k=1}^n LAT_k \left( \frac{1}{2} DEP_k + \sum_{j=1}^{k-1} DEP_j \right)$$

where:

$N, E$  = northing, easting of a point.

$k$  = a current point.

$n$  = number of points in the survey.

$AZ$  = azimuth of a course.

$HD$  = horizontal distance.

$SD$  = slope distance.

$ZA$  = zenith angle.

## Program Listing.

F01 LBL F	S05 SIN
F02 SF 0	S06 x
F03 CLVARS	S07 STO H
F04 INPUT N	S08 CF 0
F05 INPUT E	Checksum = 53D5
F06 INPUT F	H01 LBL H
F07 →HR	H02 FS? 0
F08 100	H03 INPUT H
F09 +	H04 STO+ T
F10 STO F	H05 RCL F
F11 STOP	H06 SF 0
Checksum = 1CC1	H07 x<>y
A01 LBL A	H08 θ,r→y,x
A02 INPUT A	H09 STO+ N
A03 →HR	H10 x<>y
A04 100	H11 STO+ E
A05 +	H12 STO X
A06 STO+ F	H13 2
A07 STOP	H14 ÷
Checksum = D137	H15 RCL+ K
D01 LBL D	H16 x
D02 INPUT D	H17 STO+ R
D03 →HR	H18 RCL X
D04 STO+ F	H19 STO+ K
D05 STOP	H20 VIEW N
Checksum = 025F	H21 VIEW E
S01 LBL S	H22 VIEW T
S02 INPUT S	H23 VIEW R
S03 INPUT Z	H24 RTN
S04 →HR	Checksum = B55D

**Flags Used.** Flag 0.

**Memory Required.** 98.5 bytes.

## Remarks.

- This program clears all variables stored in Continuous Memory.
- Right angles and deflections are positive; left angles and deflections are negative.
- This program requires the calculator to be set to *degrees* mode; angular inputs must be in degrees-minutes-seconds (D.MS) format.
- The program uses zenith angles to calculate the horizontal distance from slope distance. If you are using vertical angles rather than zenith angles, convert the vertical angle to a zenith angle by using:

$$\text{zenith angle} = 90^\circ - \text{vertical angle}.$$

(Remember to convert D.MS input to decimal degrees before subtracting from 90.)

## Program Instructions.

1. Key in the program listing and press  $\boxed{C}$  when finished.
2. Press  $\boxed{XEQ}$  F.
3. Key in N and press  $\boxed{R/S}$  ; key in E and press  $\boxed{R/S}$  ; key in F and press  $\boxed{R/S}$  .
4. Select the appropriate routine to input the direction:
  - For an angle right or an angle left:
    - Press  $\boxed{XEQ}$  A.
    - Key in A.
    - If angle left, press  $\boxed{+/-}$  .
    - Press  $\boxed{R/S}$  .
  - For a deflection right or a deflection left:
    - Press  $\boxed{XEQ}$  D.
    - Key in D.
    - If deflection left, press  $\boxed{+/-}$  .
    - Press  $\boxed{R/S}$  .

5. Select the appropriate routine to input the distance:
  - For a horizontal distance:
    - Press  $\boxed{\text{XEQ}}$  H.
    - Key in  $A$  and press  $\boxed{\text{R/S}}$  .
    - See  $N$  and press  $\boxed{\text{R/S}}$  .
    - See  $E$ .
  - For a slope distance with zenith angle:
    - Press  $\boxed{\text{XEQ}}$  S.
    - Key in  $S$  and press  $\boxed{\text{R/S}}$  .
    - Key in  $Z$  and press  $\boxed{\text{R/S}}$  .
    - See  $N$  and press  $\boxed{\text{R/S}}$  .
    - See  $E$ .
6. When the final distance and angle have been keyed in, press  $\boxed{\text{R/S}}$  to see  $T$ .
7. Press  $\boxed{\text{R/S}}$  to see  $R$ .

### **Variables Used.**

$N$  = the northing of each point.

$E$  = the easting of each point.

$F$  = reference azimuth away from the starting point.

$A$  = angle change of direction.

$D$  = deflection change of direction.

$H$  = horizontal distance.

$S$  = slope distance.


$Z$  = zenith angle.

$T$  = total distance traversed.

$R$  = enclosed area.

$X, K$  = variables used for intermediate results.

**Example: Field Angle Traverse.** Find the coordinates of each point, the total distance, and the area enclosed for the following field:



**Keys:**

■ **MODES** {DG}

**XEQ** F

150 **R/S**

400 **R/S**

311.3955 **R/S**

**XEQ** A

113.3455 **R/S**

**XEQ** H

177.966 **R/S**

**R/S**

**XEQ** D

100.2455 **+/-** **R/S**

**XEQ** S

161.88 **R/S**

86.0139 **R/S**

**R/S**

**XEQ** A

87.3559 **R/S**

**XEQ** H

203.69 **R/S**

**R/S**

**XEQ** D

100.4559 **+/-** **R/S**

**XEQ** H

124 **R/S**

**R/S**

**R/S**

**R/S**

**Display:**

N?0.0000

E?0.0000

F?0.0000

491.6653

A?0.0000

293.5819

H?0.0000

N=224.5150

E=561.6150

D?0.0000

-100.4153

S?0.0000

Z?0.000

N=356.5285

E=468.5999

A?113.3455

267.5997

H?161.4911

N=232.3373

E=307.1498

D?-100.2455

-100.7664

H?203.6900

N=149.9048

E=399.7829

T=667.1471

R=26,558.8326

**Description:**

Sets *degrees* mode.

Inputs starting point data.

Displays  $N_2$ .

Displays  $E_2$ .

Displays  $N_3$ .

Displays  $E_3$ .

Displays  $N_4$ .

Displays  $E_4$ .

Displays  $N_1$ .

Displays  $E_1$ .

Displays total distance.

Displays area.



# 4

## Statistics

---

---

## t Statistics

This program performs t statistics calculations for either paired statistics or for two means.

### Paired t Statistics

Given a set of paired observations from two normal populations with unknown means  $\mu_1$  and  $\mu_2$ :

$$\begin{array}{c|cccc} x_i & x_1 & x_2 & \cdots & x_n \\ \hline y_i & y_1 & y_2 & \cdots & y_n \end{array}$$

let:

$$\begin{aligned} D_i &= x_i - y_i \\ \bar{D} &= \frac{1}{n} \sum_{i=1}^n D_i \\ s_D &= \left[ \frac{\sum D_i^2 - \frac{1}{n} (\sum D_i)^2}{n - 1} \right]^{\frac{1}{2}} \end{aligned}$$

The test statistic

$$t = \frac{\bar{D}}{s_D} \sqrt{n}$$

has  $n - 1$  degrees of freedom ( $df$ ) and can be used to test the null hypothesis  $H_0 : \mu_1 = \mu_2$ .

## t Statistic for Two Means

Suppose  $\{x_1, x_2, \dots, x_{n_1}\}$  and  $\{y_1, y_2, \dots, y_{n_2}\}$  are independent random samples from two normal populations having means  $\mu_1$  and  $\mu_2$  (unknown) and the same unknown variance  $\sigma^2$ .

To test the null hypothesis  $H_0 : \mu_1 - \mu_2 = d$ , where  $d$  is a given number, use the following equations:

$$\bar{x} = \frac{1}{n_1} \sum_{i=1}^{n_1} x_i$$

$$\bar{y} = \frac{1}{n_2} \sum_{i=1}^{n_2} y_i$$

$$t = \frac{\bar{x} - \bar{y} - d}{\left[ \left( \frac{1}{n_1} + \frac{1}{n_2} \right) \left( \frac{\sum x_i^2 - n_1 \bar{x}^2 + \sum y_i^2 - n_2 \bar{y}^2}{n_1 + n_2 - 2} \right) \right]^{\frac{1}{2}}}$$

You can use this  $t$  statistic, which has the  $t$  distribution with  $n_1 + n_2 - 2$  degrees of freedom ( $df$ ), to test the null hypothesis  $H_0$ .

## Program Listing.

P01 LBL P	T07 $\Sigma x^2$
P02 CL $\Sigma$	T08 STO B
P03 INPUT N	T09 n
Checksum = FCFE	T10 STO C
K01 LBL K	T11 CL $\Sigma$
K02 INPUT X	T12 STOP
K03 INPUT Y	T13 $\bar{x}$
K04 RCL X	T14 STO J
K05 RCL- Y	T15 $\Sigma x^2$
K06 $\Sigma+$	T16 STO K
K07 DSE N	T17 n
K08 GT0 K	T18 STO L
K09 $\bar{x}$	T19 RCL B
K10 STO D	T20 RCL A
K11 $s_x$	T21 $x^2$
K12 STO S	T22 RCL $\times$ C
K13 $\div$	T23 -
K14 n	T24 RCL+ K
K15 SQRT	T25 RCL J
K16 $\times$	T26 $x^2$
K17 STO T	T27 RCL $\times$ L
K18 n	T28 -
K19 1	T29 RCL C
K20 -	T30 RCL+ L
K21 STO F	T31 2
K22 VIEW D	T32 -
K23 VIEW S	T33 STO F
K24 VIEW T	T34 $\div$
K25 VIEW F	T35 SQRT
K26 RTN	T36 $1/x$
Checksum = 16B7	T37 RCL A
T01 LBL T	T38 RCL- J
T02 INPUT H	T39 RCL- H
T03 CL $\Sigma$	T40 $\times$
T04 STOP	T41 RCL C
T05 $\bar{x}$	T42 $1/x$
T06 STO A	T43 RCL L

T44  $1/x$   
T45 +  
T46 SQRT  
T47  $\div$   
T48 STO T

T49 VIEW T  
T50 VIEW F  
T51 RTN  
Checksum = A79C

**Flags Used.** None.

**Memory Required.** 120 bytes.

**Remarks.**

- *This program clears all statistical data stored in Continuous Memory.*
- The two routines are independent of each other.

**Program Instructions.**

1. Key in the programs to be used; press **[C]** when finished.
2. For paired t statistics:
  - Press **[XEQ] P**.
  - Key in  $N$  and press **[R/S]**.
  - Key in each  $X$  and press **[R/S]**; key in the corresponding  $Y$  and press **[R/S]**.
  - See the results as they are displayed. Press **[R/S]** to display the next result.
3. For t statistics of two means:
  - Press **[XEQ] T**.
  - Key in  $H$  and press **[R/S]**.
  - Key in each  $x$ -value and press **[ $\Sigma+$ ]**.
  - When all of the  $x$ -values have been entered, press **[R/S]**.
  - Key in each  $y$ -value and press **[ $\Sigma+$ ]**.
  - When all of the  $y$ -values have been entered, press **[R/S]** to calculate the test statistic  $T$ .
  - Press **[R/S]** to calculate the degrees of freedom.

**Variables Used.**

$X = x$ -value of a pair of observations.

$Y = y$ -value of a pair of observations.

$N =$  number of paired values.

$D =$  average difference,  $\bar{D}$ .

$S =$  standard deviation,  $s_D$ .

$F =$  number of degrees of freedom,  $df$ .

$H =$  null hypothesis difference,  $d$ .

$T =$  test statistic.

$A, B, C, J, K, L =$  variables used for intermediate results.

**Example 1: Paired Observations.** Calculate the test statistic and degrees of freedom of the following data pairs for the null hypothesis  $H_0: \mu_1 = \mu_2$ .

<b>x</b>	15	16.9	15.3	17	19.1	15.3
<b>y</b>	18	19.3	17	20.3	19.7	18

**Keys:**

**Display:**

**Description:**

XEQ P

N?value

6  R/S

X?value

Inputs values.

15  R/S

Y?value

18  R/S

X?15.0000

16.9  R/S

Y?18.0000

19.3  R/S

X?16.9000

15.3  R/S

Y?19.3000

17  R/S

X?15.3000

17  R/S

Y?17.0000

20.3  R/S

X?17.0000

19.1  R/S

Y?20.3000

19.7  R/S

X?19.1000

15.3	$\boxed{\text{R/S}}$	Y?	19.7000	
18	$\boxed{\text{R/S}}$	D=	-2.2833	Displays $\bar{D}$ .
	$\boxed{\text{R/S}}$	S=	0.9908	Displays $s_D$ .
	$\boxed{\text{R/S}}$	T=	-5.6450	Displays $t$ .
	$\boxed{\text{R/S}}$	F=	5.0000	Displays $df$ .

**Example 2: Two Means.** Calculate the test statistic and degrees of freedom of the following data for the null hypothesis  $H_0: \mu_1 = \mu_2$ .

<b>x</b>	86	109	112	91	103	121	107	100	97
<b>y</b>	93	101	111	117	105	97	99		

Keys:	Display:	Description:
$\boxed{\text{XEQ}}$ T	H?value	
0 $\boxed{\text{R/S}}$	0.0000	
86 $\boxed{\Sigma+}$	1.0000	Inputs x-values.
109 $\boxed{\Sigma+}$	2.0000	
112 $\boxed{\Sigma+}$	3.0000	
91 $\boxed{\Sigma+}$	4.0000	
103 $\boxed{\Sigma+}$	5.0000	
121 $\boxed{\Sigma+}$	6.0000	
107 $\boxed{\Sigma+}$	7.0000	
100 $\boxed{\Sigma+}$	8.0000	
97 $\boxed{\Sigma+}$	9.0000	
$\boxed{\text{R/S}}$	9.0000	
93 $\boxed{\Sigma+}$	1.0000	Inputs y-values.
101 $\boxed{\Sigma+}$	2.0000	
111 $\boxed{\Sigma+}$	3.0000	
117 $\boxed{\Sigma+}$	4.0000	
105 $\boxed{\Sigma+}$	5.0000	
97 $\boxed{\Sigma+}$	6.0000	
99 $\boxed{\Sigma+}$	7.0000	
$\boxed{\text{R/S}}$	T = -0.0801	Displays $t$ .
$\boxed{\text{R/S}}$	F = 14.0000	Displays $df$ .

---

## Chi-Square Evaluation

This program calculates the value of the  $\chi^2$  statistic for the goodness of fit test using the equation:

$$\chi^2 = \sum_{i=1}^n \frac{(O_i - E_i)^2}{E_i} \quad (df = n - 1)$$

where:

$O_i$  = the observed frequency.

$E_i$  = the expected frequency.

If the expected values are equal:

$$E = E_i = \frac{\Sigma O_i}{n} \quad \text{for all } i$$

then:

$$\chi^2 = \frac{n \Sigma O_i^2}{\Sigma O_i} - \Sigma O_i$$

### Program Listing.

```
U01 LBL U           D11 VIEW C
U02 0               D12 RTN
U03 STO C           Checksum = 989C
U04 INPUT N         E01 LBL E
Checksum = 05C2     E02 CLΣ
D01 LBL D           E03 INPUT N
D02 INPUT O         Checksum = 222E
D03 INPUT E         K01 LBL K
D04 RCL O           K02 INPUT O
D05 RCL- E         K03 Σ+
D06 x²              K04 DSE N
D07 RCL÷ E         K05 GTO K
D08 STO+ C         K06 n
D09 DSE N          K07 Σx²
D10 GTO D          K08 x
```



K09 $\Sigma x$	K15 $\bar{x}$
K10 $\div$	K16 STO E
K11 LAST $x$	K17 VIEW E
K12 -	K18 RTN
K13 STO C	Checksum = 23FF
K14 VIEW C	

**Flags Used.** None.

**Memory Required.** 55.5 bytes.

**Remarks.**

- *This program clears all statistical data stored in Continuous Memory.*
- The two routines (the unequal case and the equal case) are independent of each other.

**Program Instructions.**

1. Key in the program listing; press [C] when finished.
2. Select the appropriate routine:
  - Press [XEQ] U if the expected values are unequal.
  - Press [XEQ] E if the expected values are equal.
3. Key in the variables at each prompt and press [R/S] .
4. After all data is input, the  $\chi^2$  value is calculated and displayed.
5. For a new case, go to step 2.

**Variables Used.**

$O$  = observed frequency.

$E$  = expected frequency.

$C = \chi^2$ .

$N$  = number of data pairs (unequal case) or values (equal case).

**Example 1: Unequal Expected Frequencies.** Find the  $\chi^2$  statistic for the goodness of fit for the following data set:

<b>Observed</b>	8	50	47	56	5	14
<b>Expected</b>	9.6	46.75	51.85	54.4	8.25	9.15

**Keys:**

**Display:**

**Description:**

$\boxed{\text{XEQ}}$  U

N?value

6  $\boxed{\text{R/S}}$

O?value

Inputs values.

8  $\boxed{\text{R/S}}$

E?value

9.6  $\boxed{\text{R/S}}$

O?8.0000

50  $\boxed{\text{R/S}}$

E?9.6000

46.75  $\boxed{\text{R/S}}$

O?50.0000

47  $\boxed{\text{R/S}}$

E?46.7500

51.85  $\boxed{\text{R/S}}$

O?47.0000

56  $\boxed{\text{R/S}}$

E?51.8500

54.4  $\boxed{\text{R/S}}$

O?56.0000

5  $\boxed{\text{R/S}}$

E?54.4000

8.25  $\boxed{\text{R/S}}$

O?5.0000

14  $\boxed{\text{R/S}}$

E?8.2500

9.15  $\boxed{\text{R/S}}$

C=4.8444

Displays  $\chi^2$ .

**Example 2: Equal Expected Frequencies.** The following table shows the frequencies observed in tossing a die 120 times.  $\chi^2$  can be used to test if the die is fair ( $df=5$ ). Assume that the expected frequencies are equal.

<b>i</b>	1	2	3	4	5	6
<b>Observed</b>	25	17	15	23	24	16

**Keys:**

XEQ E

6  R/S

25  R/S

17  R/S

15  R/S

23  R/S

24  R/S

16  R/S

**Display:**

N?value

O?value

O?25.0000

O?17.0000

O?15.0000

O?23.0000

O?24.0000

C=5.0000

**Description:**

Inputs values.

Calculates and displays  $\chi^2$ .

The value of  $\chi^2$  for  $df=5$  and 5% significance\* is 11.070. Since 5.00 is less than 11.070, no statistically significant differences exist between the observed and expected frequencies.

---

\* See page 438 of J.E. Freund's *Mathematical Statistics*, 2nd edition.


---

# F Distribution

This program evaluates the integral of the F distribution:

$$Q(x) = \int_x^{\infty} \frac{\Gamma\left(\frac{\nu_1 + \nu_2}{2}\right) y^{\frac{\nu_1}{2} - 1} \left(\frac{\nu_1}{\nu_2}\right)^{\frac{\nu_1}{2}}}{\Gamma\left(\frac{\nu_1}{2}\right) \Gamma\left(\frac{\nu_2}{2}\right) \left(1 + \frac{\nu_1}{\nu_2} y\right)^{\frac{\nu_1 + \nu_2}{2}}} dy$$

where  $x > 0$  and  $\nu_1$  and  $\nu_2$  are the degrees of freedom, provided either  $\nu_1$  or  $\nu_2$  is even and both are greater than two.



The following series are used to evaluate the integral.

If  $\nu_1$  is even:

$$Q(x) = t^{\frac{\nu_2}{2}} \left[ 1 + \frac{\nu_2}{2}(1-t) + \frac{\nu_2(\nu_2+2)}{2 \cdot 4}(1-t)^2 + \dots + \frac{\nu_2(\nu_2+2) \cdots (\nu_2 + \nu_1 - 4)}{2 \cdot 4 \cdots (\nu_1 - 2)} (1-t)^{\frac{\nu_1-2}{2}} \right]$$

If  $\nu_2$  is even:

$$Q(x) = 1 - (1-t)^{\frac{\nu_1}{2}} \left[ 1 + \frac{\nu_1}{2}t + \frac{\nu_1(\nu_1+2)}{2 \cdot 4}t^2 + \dots + \frac{\nu_1(\nu_1+2) \cdots (\nu_2 + \nu_1 - 4)}{2 \cdot 4 \cdots (\nu_2 - 2)} t^{\frac{\nu_2-2}{2}} \right]$$

where:

$$t = \frac{\nu_2}{\nu_2 + \nu_1 x}$$

### Program Listing.

F01 LBL F	E02 SF 0
F02 CF 0	E03 RCL A
F03 INPUT A	E04 RCL B
F04 INPUT B	E05 STO A
F05 INPUT X	E06 x<>y
F06 RCL× A	E07 STO B
F07 RCL+ B	E08 1
F08 RCL÷ B	E09 RCL- T
F09 1/x	E10 STO T
F10 STO T	Checksum = 2FFE
F11 RTN	D01 LBL D
Checksum = 4D69	D02 0
E01 LBL E	D03 STO I

D04 1	P15 RCL+ B
D05 STO K	P16 ×
D06 STO H	P17 STO K
D07 RCL A	P18 STO+ H
D08 2	P19 RCL N
D09 -	P20 RCL- I
D10 2	P21 ≠0?
D11 ÷	P22 GT0 P
D12 STO N	P23 RCL B
Checksum = 9A01	P24 2
P01 LBL P	P25 ÷
P02 1	P26 RCL T
P03 STO+ I	P27 x<>y
P04 RCL- T	P28 y <sup>x</sup>
P05 RCL× K	P29 RCL× H
P06 2	P30 1
P07 RCL I	P31 x<>y
P08 y <sup>x</sup>	P32 FS? 0
P09 ÷	P33 -
P10 RCL I	P34 STO Q
P11 1	P35 VIEW Q
P12 -	P36 RTN
P13 2	Checksum = 5194
P14 ×	

**Flags Used.** Flag 0.

**Memory Required.** 103.5 bytes.

### Program Instructions.

1. Key in the program listing; press **[C]** when finished.
2. Press **[XEQ] F**.
3. Key in the variables as they are prompted for, pressing **[R/S]** after each entry.
4. Select the appropriate routine to calculate  $Q(x)$ :
  - Press **[XEQ] E** if  $\nu_1$  is odd and  $\nu_2$  is even.
  - Press **[XEQ] D** if  $\nu_1$  is even and  $\nu_2$  is odd.
  - Press **[XEQ] D** if both  $\nu_1$  and  $\nu_2$  are even.
5. For a new case, go to step 2.

## Variables Used.

$$A = \nu_1.$$

$$B = \nu_2.$$

$$X = x.$$

$$Q = Q(x).$$

$T, I, K, H, N$  = variables used for intermediate results.

**Example 1.** Calculate  $Q(3.92)$ , where  $\nu_1 = 9$  and  $\nu_2 = 6$ .

Keys:	Display:	Description:
$\boxed{\text{XEQ}}$ F	$A?value$	
9 $\boxed{\text{R/S}}$	$B?value$	Inputs values.
6 $\boxed{\text{R/S}}$	$X?value$	
3.92 $\boxed{\text{R/S}}$	0.1453	
$\boxed{\text{XEQ}}$ E	Q=0.0552	Displays $Q(3.92)$ .

**Example 2.** Calculate  $Q(1.85)$ , where  $\nu_1 = 4$  and  $\nu_2 = 16$ .

Keys:	Display:	Description:
$\boxed{\text{XEQ}}$ F	$A?value$	
4 $\boxed{\text{R/S}}$	$B?value$	Inputs values.
16 $\boxed{\text{R/S}}$	$X?value$	
1.85 $\boxed{\text{R/S}}$	0.6838	
$\boxed{\text{XEQ}}$ D	Q=0.1687	Displays $Q(1.85)$ .

## Analysis of Variance (One Way)

One way analysis of variance tests the difference between the population means of  $k$  treatment groups. Group  $i$  ( $i = 1, 2, \dots, k$ ) has  $n_i$  observations. Treatment groups may have equal or unequal numbers of observations.

$$\text{Sum}_i = \sum \text{of observations in treatment group } i = \sum_{j=1}^{n_i} x_{ij}$$

$$\text{Total } SS = \sum_{i=1}^k \sum_{j=1}^{n_i} x_{ij}^2 - \frac{\left( \sum_{i=1}^k \sum_{j=1}^{n_i} x_{ij} \right)^2}{\sum_{i=1}^k n_i}$$

$$\text{Treat } SS = \sum_{i=1}^k \left[ \frac{\left( \sum_{j=1}^{n_i} x_{ij} \right)^2}{n_i} \right] - \frac{\left( \sum_{i=1}^k \sum_{j=1}^{n_i} x_{ij} \right)^2}{\sum_{i=1}^k n_i}$$

$$\text{Error } SS = \text{Total } SS - \text{Treat } SS$$

$$df_1 = \text{Treat } df = k - 1$$

$$df_2 = \text{Error } df = \sum_{i=1}^k n_i - k$$

$$\text{Treat } MS = \frac{\text{Treat } SS}{\text{Treat } df}$$

$$\text{Error } MS = \frac{\text{Error } SS}{\text{Error } df}$$

$$F = \frac{\text{Treat } MS}{\text{Error } MS} \quad \left( \text{with } k - 1 \text{ and } \sum_{i=1}^k n_i - k \text{ degrees of freedom} \right)$$



## Program Listing.

A01 LBL A	L22 $x^2$
A02 CLVARS	L23 RCL÷ N
A03 INPUT K	L24 -
A04 STO J	L25 LASTx
Checksum = B956	L26 +/-
L01 LBL L	L27 RCL+ C
L02 CLx	L28 -
L03 CLΣ	L29 LASTx
L04 STOP	L30 ÷
L05 Σx	L31 $1/x$
L06 STO S	L32 RCL K
L07 VIEW S	L33 1
L08 STO+ A	L34 -
L09 Σx <sup>2</sup>	L35 STO D
L10 STO+ B	L36 VIEW D
L11 n	L37 ÷
L12 STO+ N	L38 RCL N
L13 Σx	L39 RCL- K
L14 $x^2$	L40 STO D
L15 n	L41 VIEW D
L16 ÷	L42 ×
L17 STO+ C	L43 STO F
L18 DSE J	L44 VIEW F
L19 GT0 L	L45 RTN
L20 RCL B	Checksum = 0443
L21 RCL A	

**Flags Used.** None.

**Memory Required.** 73.5 bytes.

**Remarks.** *This program clears all variables and statistical data stored in Continuous Memory.*

### Program Instructions.

1. Key in the program listing; press  $\boxed{C}$  when finished.
2. Press  $\boxed{XEQ}$  A.
3. Key in  $K$  (the number of treatment groups) and press  $\boxed{R/S}$ .
4. Key in each observation and press  $\boxed{\Sigma+}$  .
5. Press  $\boxed{R/S}$  when all of the observations in the treatment group have been entered.
6. See the sum and press  $\boxed{R/S}$  .
7. See the treatment degrees of freedom ( $df_1$ ) and press  $\boxed{R/S}$  .
8. See the error degrees of freedom ( $df_2$ ) and press  $\boxed{R/S}$  .
9. See the F ratio ( $F$ ).

### Variables Used.

$K$  = number of treatment groups.

$S$  = sum of observations in a treatment group.

$D = df_1$  and  $df_2$  .

$F$  = F ratio.

$J, A, B, N, C$  = variables used for intermediate results.

**Example.** Find  $\text{Sum}_1$ ,  $\text{Sum}_2$ ,  $\text{Sum}_3$ ,  $df_1$ ,  $df_2$ , and  $F$  for the following:

$i \backslash j$	1	2	3	4	5	6
1	10	13	12	15	17	
2	8	10	12	13	11	9
3	8	9	12	7		

**Keys:**

**Display:**

**Description:**

$\boxed{\text{XEQ}}$  A

K?0.0000

Prompts for number of treatment groups.

3  $\boxed{\text{R/S}}$

0.0000

10  $\boxed{\Sigma+}$

1.0000

Inputs observations in first treatment group.

13  $\boxed{\Sigma+}$

2.0000

12  $\boxed{\Sigma+}$

3.0000

15  $\boxed{\Sigma+}$

4.0000

17  $\boxed{\Sigma+}$

5.0000

$\boxed{\text{R/S}}$

S=67.0000

Displays  $\text{Sum}_1$ .

$\boxed{\text{R/S}}$

0.0000

Inputs observations in second treatment group.

8  $\boxed{\Sigma+}$

1.0000

10  $\boxed{\Sigma+}$

2.0000

12  $\boxed{\Sigma+}$

3.0000

13  $\boxed{\Sigma+}$

4.0000

11  $\boxed{\Sigma+}$

5.0000

9  $\boxed{\Sigma+}$

6.0000

$\boxed{\text{R/S}}$

S=63.0000

Displays  $\text{Sum}_2$ .

$\boxed{\text{R/S}}$

0.0000

Inputs observations in third observation group.

8  $\boxed{\Sigma+}$

1.0000

9  $\boxed{\Sigma+}$

2.0000

12  $\boxed{\Sigma+}$

3.0000

7  $\boxed{\Sigma+}$

4.0000

$\boxed{\text{R/S}}$

S=36.0000

Displays  $\text{Sum}_3$ .

$\boxed{\text{R/S}}$

D=2.0000

Displays  $df_1$ .

$\boxed{\text{R/S}}$

D=12.0000

Displays  $df_2$ .

$\boxed{\text{R/S}}$

F=4.5700

Displays  $F$ .

---

## Binomial Distribution

This program calculates the probability of a value falling within a specified range of values, that is, the cumulative distribution  $\sum_{x=B}^A p(x)$ , in a binomial distribution. It also calculates the mean, the variance, and the standard deviation of the distribution, and can be used to find the value of each term in the distribution.

$$p(x) = \binom{n}{x} r^x (1-r)^{n-x}$$

where  $x = 0, 1, 2, \dots$  and  $r < 1$ .

### Program Listing.

B01 LBL B	Y12 $y^x$
B02 CLVARs	Y13 LASTx
B03 INPUT N	Y14 RCL N
B04 INPUT R	Y15 $x \langle \rangle y$
B05 INPUT B	Y16 Cn,r
B06 INPUT A	Y17 $\times$
B07 1,000	Y18 $\times$
B08 $\div$	Y19 FS? 0
B09 STO+ B	Y20 STOP
Checksum = 6A51	Y21 STO+ P
Y01 LBL Y	Y22 ISG B
Y02 1	Y23 GTO Y
Y03 RCL- R	Y24 VIEW P
Y04 RCL N	Y25 RCL N
Y05 RCL B	Y26 RCL $\times$ R
Y06 IP	Y27 STO M
Y07 -	Y28 1
Y08 $y^x$	Y29 RCL- R
Y09 RCL R	Y30 $\times$
Y10 RCL B	Y31 STO V
Y11 IP	Y32 SQRT

Y33 STO S  
Y34 VIEW M  
Y35 VIEW V

Y36 VIEW S  
Y37 RTN  
Checksum = 2C13

**Flags Used.** Flag 0.

**Memory Required.** 77 bytes.

### Remarks.

- *This program clears all variables stored in Continuous Memory.*
- The upper and lower limits of the range are inclusive ( $B \leq x \leq A$ ). If the limits are exclusive or noninteger values, round the lower limit to the next highest integer and the upper limit to the next lowest integer.
- The limits  $A$  and  $B$  have no effect on the mean, variance, and standard deviation.
- An invalid data error will result if  $B < 0$  or if  $A > n$ .

### Program Instructions.

1. Key in the program listing; press **[C]** when finished.
2. Press **[XEQ] B**.
3. Key in the variables at each prompt and press **[R/S]**.
4. Optional: To see each term of the distribution, set flag 0; press **[R/S]** to continue execution.
5. See each result as it is displayed and press **[R/S]**.
6. For a new case, go to step 2.

## Variables Used.

$N$  = number of events.

$R$  = probability of the occurrence of a single event.

$A$  = upper limit of the range.

$B$  = lower limit of the range.

$P$  = probability of a value falling in the range.

$M$  = mean.

$V$  = variance.

$S$  = standard deviation.

**Example 1.** A fair coin ( $r = 0.5$ ) is tossed 10 times. What is the probability that at least seven heads will occur? Find the mean, variance, and standard deviation.

### Keys:

■ [FLAGS] {CF} 0

[XEQ] B

10 [R/S]

.5 [R/S]

7 [R/S]

10 [R/S]

[R/S]

[R/S]

[R/S]

### Display:

*value*

N?0.0000

R?0.0000

B?0.0000

A?0.0000

P=0.1719

M=5.0000

V=2.5000

S=1.5811

### Description:

Clears flag 0.

Inputs values.

Displays probability.

Displays mean.

Displays variance.

Displays standard deviation.

**Example 2.** Find the terms of the binomial distribution with  $n = 5$  and  $r = 0.75$ .

<b>Keys:</b>	<b>Display:</b>	<b>Description:</b>
<b>■</b> <b>FLAGS</b> {SF} 0	<i>value</i>	Sets flag to display each term of distribution.
<b>XEQ</b> B	N?0.0000	
5 <b>R/S</b>	R?0.0000	Inputs values.
.75 <b>R/S</b>	B?0.0000	
<b>R/S</b>	A?0.0000	
5 <b>R/S</b>	0.0010	Displays $p(0)$ .
<b>R/S</b>	0.0146	Displays $p(1)$ .
<b>R/S</b>	0.0879	Displays $p(2)$ .
<b>R/S</b>	0.2637	Displays $p(3)$ .
<b>R/S</b>	0.3955	Displays $p(4)$ .
<b>R/S</b>	0.2373	Displays $p(5)$ .

---

## Poisson Distribution

This program calculates the probability of a value falling within a specified range of values, that is, the cumulative distribution  $\sum_{x=B}^A p(x)$ , in a Poisson distribution. It also calculates the mean, the variance, and the standard deviation of the distribution, and can be used to find the value of each term in the distribution.

$$p(x) = \frac{e^{-\lambda} \lambda^x}{x!}$$

where  $x = 0, 1, 2, \dots$  and  $\lambda > 0$

### Program Listing.

P01 LBL P	X11 e <sup>x</sup>
P02 CLVARS	X12 x
P03 INPUT L	X13 FS? 0
P04 INPUT B	X14 STOP
P05 INPUT A	X15 STO+ P
P06 1,000	X16 ISG B
P07 ÷	X17 GTO X
P08 STO+ B	X18 VIEW P
Checksum = 0BD9	X19 RCL L
X01 LBL X	X20 STO M
X02 RCL L	X21 STO V
X03 RCL B	X22 SQRT
X04 IP	X23 STO S
X05 y <sup>x</sup>	X24 VIEW M
X06 LASTx	X25 VIEW V
X07 x!	X26 VIEW S
X08 ÷	X27 RTN
X09 RCL L	Checksum = B9BA
X10 +/-	



**Flags Used.** Flag 0.

**Memory Required.** 60.5 bytes.

**Remarks.**

- *This program clears all variables stored in Continuous Memory.*
- The upper and lower limits of the range are inclusive ( $B \leq x \leq A$ ). If the limits are exclusive or noninteger values, round the lower limit to the next highest integer and the upper limit to the next lowest integer.
- The limits  $A$  and  $B$  have no effect on the mean, variance, and standard deviation.  $A$  must be  $\leq 999$ ,  $B$  must be  $\geq 0$ .

**Program Instructions.**

1. Key in the program listing; press [C] when finished.
2. Press [XEQ] P.
3. Key in the variables at each prompt and press [R/S] .
4. Optional: To see each term of the distribution, set flag 0; press [R/S] to continue execution.
5. See each result as it is displayed and press [R/S] .
6. For a new case, go to step 2.

**Variables Used.**

$L = \lambda$ .

$A$  = upper limit of the range.

$B$  = lower limit of the range.

$P$  = probability of a value falling in the range.

$M$  = mean.

$V$  = variance.

$S$  = standard deviation.

**Example 1.** For a Poisson distribution with  $\lambda = 2$ , find the probability that  $0 < x < 2.5$ ; also find the mean, the variance, and the standard deviation. (Remember that the Poisson distribution deals only with integers. Therefore, the only  $x$  values in this range are 1 and 2.)

Keys:	Display:	Description:
■ [FLAGS] {CF} 0	<i>value</i>	Clears flag 0.
[XEQ] P	L?0.0000	
2 [R/S]	B?0.0000	Inputs values.
1 [R/S]	A?0.0000	
2 [R/S]	P=0.5413	Displays probability.
[R/S]	M=2.0000	Displays mean.
[R/S]	V=2.0000	Displays variance.
[R/S]	S=1.4142	Displays standard deviation.

**Example 2.** Find the first six terms (from  $x = 0$  to  $x = 5$ ) of the Poisson distribution with  $\lambda = 3$ .

Keys:	Display:	Description:
■ [FLAGS] {SF} 0	<i>value</i>	Sets flag to display each term of distribution.
[XEQ] P	L?0.0000	
3 [R/S]	B?0.0000	Inputs values.
[R/S]	A?0.0000	
5 [R/S]	0.0498	Displays $p(0)$ .
[R/S]	0.1494	Displays $p(1)$ .
[R/S]	0.2240	Displays $p(2)$ .
[R/S]	0.2240	Displays $p(3)$ .
[R/S]	0.1680	Displays $p(4)$ .
[R/S]	0.1008	Displays $p(5)$ .


# Mathematics

---

---

# Triangle Solutions

This program may be used to find the sides, angles, and area of a plane triangle.



In general, the specifications of any three of the six parameters of a triangle (three sides and three angles) is sufficient to define a triangle (the exception is that three angles will not define a triangle).

This program will handle all five cases:

- Three sides (SSS).
- Two angles and the included side (ASA).
- Two angles and the adjacent side (SAA).
- Two sides and the included angle (SAS).
- Two sides and the adjacent angle (SSA).

The last case listed (SSA) may result in two solutions to the triangle. This program will calculate both solutions.

If the three known input values are selected in a clockwise order around the triangle, the output values will also follow a clockwise order.

## Program Listing.

```
A01 LBL A
A02 INPUT A
A03 INPUT C
A04 INPUT E
A05 RCL A
A06 RCL C
A07 y, x→θ, r
A08 x²
A09 RCL E
A10 x²
A11 -
A12 2
A13 RCL× A
A14 RCL× C
A15 ÷
A16 ACOS
A17 STO B
A18 GTO K
Checksum = 9E72
B01 LBL B
B02 INPUT F
B03 INPUT A
B04 INPUT B
B05 RCL F
B06 SIN
B07 RCL B
B08 RCL+ F
B09 SIN
B10 ÷
B11 RCL× A
B12 STO C
B13 GTO K
Checksum = 5DAB
C01 LBL C
C02 INPUT A
C03 INPUT B
C04 INPUT D
C05 RCL+ B
C06 SIN
C07 RCL D
C08 SIN
C09 ÷
C10 RCL× A
C11 STO C
C12 GTO K
Checksum = D18A
E01 LBL E
E02 CF 2
E03 INPUT A
E04 INPUT C
E05 RCL A
E06 RCL C
E07 x>y?
E08 SF 2
E09 INPUT D
E10 SIN
E11 RCL÷ A
E12 ×
E13 ASIN
E14 RCL+ D
E15 XEQ Z
E16 STO B
E17 XEQ K
E18 RCL F
E19 XEQ Z
E20 STO F
E21 RCL+ D
E22 XEQ Z
E23 STO B
E24 GTO K
Checksum = FB3E
D01 LBL D
D02 INPUT A
D03 INPUT B
```

D04 INPUT C	K17 2
Checksum = 2A67	K18 ÷
K01 LBL K	K19 STO G
K02 RCL B	K20 VIEW A
K03 RCL A	K21 VIEW B
K04 $\theta, r \rightarrow y, x$	K22 VIEW C
K05 RCL- C	K23 VIEW D
K06 +/-	K24 VIEW E
K07 $y, x \rightarrow \theta, r$	K25 VIEW F
K08 STO E	K26 VIEW G
K09 $x \langle \rangle y$	K27 RTN
K10 STO D	Checksum = 80F0
K11 RCL+ B	Z01 LBL Z
K12 XEQ Z	Z02 COS
K13 STO F	Z03 +/-
K14 SIN	Z04 ACOS
K15 ×	Z05 RTN
K16 RCL× A	Checksum = 929B

**Flags Used.** Flag 2.

**Memory Required.** 154.5 bytes.

### Remarks.

- Angles must be consistent with the angular mode currently set in the calculator.
- Routines A through E are independent of each other. Therefore, key in only those routines that will be used. Routines K and Z must be keyed in to use any of the five routines.
- The triangle notation used by this program is *not* consistent with standard triangle notation; in other words,  $A_1$  is not opposite  $S_1$ .
- The accuracy of the solution decreases for triangles containing extremely small angles.

## Program Instructions.

1. Key in the programs to be used; press **[C]** when finished.
2. Select the appropriate routine:
  - Press **[XEQ]** A if three sides are known (SSS).
  - Press **[XEQ]** B if two angles and an included side are known (ASA).
  - Press **[XEQ]** C if two angles and an adjacent side are known (SAA).
  - Press **[XEQ]** D if two sides and an included angle are known (SAS).
  - Press **[XEQ]** E if two sides and an adjacent angle are known (SSA).
3. Key in the variables at each prompt and press **[R/S]** .
4. See each result as it is displayed. Press **[R/S]** for subsequent results.
5. If flag 2 is displayed on the calculator screen while executing routine E, a second possible solution exists. Press **[R/S]** and return to step 4 to see the second set of results.

## Variables Used.

$A$  = side<sub>1</sub>.

$B$  = angle<sub>1</sub>.

$C$  = side<sub>2</sub>.

$D$  = angle<sub>2</sub>.

$E$  = side<sub>3</sub>.

$F$  = angle<sub>3</sub>.

$G$  = the area.

**Example 1: Three Known Sides.** A farmer uses three sections of straight fence to enclose a field. The lengths are 100 feet, 120 feet, and 150 feet. Find the area enclosed and the angles formed.

**Keys:**

**Display:**

**Description:**

MODES {D/G}

XEQ A

A?value

Sets *degrees* mode.

100 R/S

C?value

Inputs lengths.

120 R/S

E?value

150 R/S

A=100.0000

Displays  $S_1$ .

R/S

B=85.4593

Displays  $A_1$ .

R/S

C=120.0000

Displays  $S_2$ .

R/S

D=41.6497

Displays  $A_2$ .

R/S

E=150.0000

Displays  $S_3$ .

R/S

F=52.8910

Displays  $A_3$ .

R/S

G=5,981.1684


Displays area.

**Example 2: Two Possible Solutions.** Given two sides and a nonincluded angle, solve for the triangle.

$$S_1 = 22.5$$

$$S_2 = 37.5$$

$$A_2 = 31.3^\circ$$





**Keys:**

**MODES** {D/G}

**XEQ**E

22.5 **R/S**

37.5 **R/S**

31.3 **R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**R/S**

**Display:**

A?value

C?value

D?value

A=22.5000

B=88.7184

C=37.5000

D=31.3000

E=43.2984

F=59.9816

G=421.7695

A=22.5000

B=28.6816

C=37.5000

D=31.3000

E=20.7860

F=120.0184

G=202.4757

**Description:**

Sets *degrees* mode.

Inputs values.  
Flag 2 is displayed.

Displays  $S_1$ .

Displays  $A_1$   
(first solution).

Displays  $S_2$ .

Displays  $A_2$ .

Displays  $S_3$ .

Displays  $A_3$ .

Displays area.

Displays  $S_1$ .  
(second solution).

Displays  $A_1$ .

Displays  $S_2$ .

Displays  $A_2$ .

Displays  $S_3$ .

Displays  $A_3$ .

Displays area.

---

## Derivative of a Function

This program calculates the derivative of a function at a given value. The function must be defined by a separate program label.

$$f'(x) \approx \frac{f(x + \delta) - f(x - \delta)}{2\delta}$$

### Program Listing.

D01 LBL D	D15 RCL X
D02 INPUT i	D16 RCL- Y
D03 INPUT X	D17 XEQ(i)
D04 ABS	D18 RCL Z
D05 $x \neq 0?$	D19 -
D06 LOG	D20 +/-
D07 IP	D21 RCL+ Y
D08 4	D22 2
D09 -	D23 $\div$
D10 $10^x$	D24 STO D
D11 STO Y	D25 VIEW D
D12 RCL+ X	D26 RTN
D13 XEQ(i)	Checksum = 20DF
D14 STO Z	

**Flags Used.** None.

**Memory Required.** 39 bytes.

**Remarks.** The program defining the function must place the value of the function in the  $X$ -register.

### Program Instructions.

1. Key in the program listing; press **[C]** when finished.
2. Key in the program that defines the function. The program should take the value in the  $X$ -register as input and leave the resulting value of the function in the  $X$ -register as output.
3. Press **[XEQ] D**. When the prompt *i?value* is displayed, specify the function by entering the number between 1 and 26 corresponding to

the program label (in other words,  $A = 1$ ,  $B = 2$ , and so on), then press **[R/S]** .

4. Key in the  $X$ -value where the function is to be evaluated and press **[R/S]** to display the derivative at that value.
5. For a new point, go to step 3.
6. For a new function, go to step 2.

### Variables Used.

$D$  = derivative of the function.

$X$  = the value at which the derivative of the function is evaluated.

$i$  = the index variable, used to specify the label of the program that defines the function.

$Y, Z$  = variables used for intermediate results.

**Example.** If  $f(x) = 3\ln(x^2 - 1)$ , find  $df/dx$  when  $x = 1.5$ .

First, enter the program for the function:

```

F01 LBL F                F05 LN
F02 x²                  F06 3
F03 1                   F07 ×
F04 -                   F08 RTN
                          Checksum = 3F9B

```

Then follow these keystrokes:

Keys:	Display:	Description:
<b>[XEQ]</b> D	$i?value$	Prompts for number corresponding to program label that defines the function.
6 <b>[R/S]</b>	$X?value$	Prompts for value where the derivative of the function is to be evaluated.
1.5 <b>[R/S]</b>	D=7.2000	Displays the derivative of the function.

---

## Linear Interpolation

Numerical relationships are often available in the form of tables. This program uses a straight line approximation to estimate a  $y$ -value given a corresponding  $x$ -value. Two pairs of  $x$ - and  $y$ -values of the relationship must be known.

$$y = \left( \frac{y_1 - y_2}{x_1 - x_2} \right) (x - x_1) + y_1$$

### Program Listing.

L01 LBL L	L11 ÷
L02 INPUT X	L12 RCL X
L03 INPUT A	L13 RCL- A
L04 INPUT B	L14 ×
L05 INPUT C	L15 RCL+ C
L06 INPUT D	L16 STO Y
L07 RCL C	L17 VIEW Y
L08 RCL- D	L18 RTN
L09 RCL A	Checksum = 96CF
L10 RCL- B	

**Flags Used.** None.

**Memory Required.** 27 bytes.

**Remarks.** The approximation is most accurate when one of the tabulated  $x$ -values is greater than the desired  $x$ -value and the other is less.

### Program Instructions.

1. Key in the program listing; press **[C]** when finished.
2. Press **[XEQ] L**.
3. Key in the variables at each prompt and press **[R/S]**.
4. See the  $y$ -value approximation.
5. For a new case, go to step 2.

## Variables Used.

$X$  = the  $x$ -value not found in the table.

$A$  = the  $x$ -value of the first pair of tabulated values.

$B$  = the  $x$ -value of the second pair.

$C$  = the  $y$ -value of the first pair of tabulated values.

$D$  = the  $y$ -value of the second pair.

$Y$  = the corresponding  $y$ -value approximation.

**Example.** The saturation pressure of steam at 110°F is 1.2763 psi, and at 120 °F it is 1.6945 psi. What is the saturation pressure when the temperature is 113°F?

### Keys:

$\boxed{\text{XEQ}}$  L

113  $\boxed{\text{R/S}}$

110  $\boxed{\text{R/S}}$

120  $\boxed{\text{R/S}}$

1.2763  $\boxed{\text{R/S}}$

1.6945  $\boxed{\text{R/S}}$

### Display:

$X?$ value

$A?$ value

$B?$ value

$C?$ value

$D?$ value

$Y = 1.4018$

### Description:

Inputs known values.

Displays approximation of the saturation pressure.

---

## Circle Determined by Three Points

This program calculates the center  $(x_0, y_0)$  and radius  $(r)$  of the circle defined by three noncollinear points.

$$r^2 = (x - x_0)^2 + (y - y_0)^2$$

### Program Listing.

C01 LBL C	C28 x⟨⟩y
C02 INPUT A	C29 STO Y
C03 INPUT B	C30 ÷
C04 INPUT C	C31 RCL Z
C05 INPUT D	C32 RCL- X
C06 INPUT E	C33 ×
C07 INPUT F	C34 RCL+ Y
C08 RCL A	C35 RCL X
C09 STO- C	C36 y, x→θ, r
C10 STO- E	C37 Z
C11 RCL B	C38 ÷
C12 STO- D	C39 STO R
C13 STO- F	C40 RCL T
C14 RCL D	C41 θ
C15 RCL C	C42 CMPLX+
C16 y, x→θ, r	C43 θ, r→y, x
C17 STO X	C44 RCL B
C18 x⟨⟩y	C45 RCL A
C19 STO T	C46 CMPLX+
C20 RCL F	C47 STO X
C21 RCL E	C48 x⟨⟩y
C22 y, x→θ, r	C49 STO Y
C23 x⟨⟩y	C50 VIEW X
C24 RCL- T	C51 VIEW Y
C25 x⟨⟩y	C52 VIEW R
C26 θ, r→y, x	C53 RTN
C27 STO Z	Checksum = A34B

**Flags Used.** None.

**Memory Required.** 79.5 bytes.

**Remarks.** A divide-by-zero error occurs if the three points are collinear. The program modifies the variables that store  $x_2$ ,  $y_2$ ,  $x_3$ , and  $y_3$ ; so if you repeat the program, you must reenter these values.

**Program Instructions.**

1. Key in the program listing; press  $\boxed{C}$  when finished.
2. Press  $\boxed{XEQ}$  C.
3. Key in the  $x$ - or  $y$ -coordinate ( $A$  through  $F$ ) at each prompt and press  $\boxed{R/S}$ .
4. After the  $y$ -coordinate of the third point is entered (with  $\boxed{R/S}$ ), the  $x$ -coordinate of the center of the circle is displayed.
5. Press  $\boxed{R/S}$  and see the  $y$ -coordinate of the center.
6. Press  $\boxed{R/S}$  and see the radius of the circle.
7. For a new case, go to step 2.

**Variables Used.**

$A$  = the  $x$ -coordinate of the first point.

$B$  = the  $y$ -coordinate of the first point.

$C$  = the  $x$ -coordinate of the second point.

$D$  = the  $y$ -coordinate of the second point.

$E$  = the  $x$ -coordinate of the third point.

$F$  = the  $y$ -coordinate of the third point.

$X$  = the  $x$ -coordinate of the center of the circle.

$Y$  = the  $y$ -coordinate of the center of the circle.

$R$  = the radius of the circle.

$T, Z$  = variables used for intermediate results.

**Example.** Find the center and radius of the circle defined by the points (1,0), (2,4.5), and (-4.4,3).

<b>Keys:</b>	<b>Display:</b>	<b>Description:</b>
$\boxed{\text{XEQ}}$ C	A?value	
1 $\boxed{\text{R/S}}$	B?value	Inputs coordinates.
0 $\boxed{\text{R/S}}$	C?value	
2 $\boxed{\text{R/S}}$	D?value	
4.5 $\boxed{\text{R/S}}$	E?value	
4.4 $\boxed{+/-}$ $\boxed{\text{R/S}}$	F?value	
3 $\boxed{\text{R/S}}$	X= -0.9775	Displays the x-coordinate of the center.
$\boxed{\text{R/S}}$	Y=2.8005	Displays the y-coordinate of the center.
$\boxed{\text{R/S}}$	R=3.4283	Displays the radius.













# Step-by-Step Solutions for Your HP-32S Calculator

---

*Engineering Applications* contains a variety of programs, examples, and solutions to show how you can easily solve your engineering problems.

■ **Electrical Engineering**

Reactance Chart • Impedance of a Ladder Network • Smith Chart Conversions • Transistor Amplifier Performance

■ **Mechanical Engineering**

Black Body Thermal Radiation • Ideal Gas Equation • Conduit Flow • Static Equivalent at a Point • Composite Section Properties • Soderberg's Equation for Fatigue

■ **Civil Engineering**

Mohr's Circle for Stress • Field Angle Traverse

■ **Statistics**

t Statistics • Chi-Square Evaluation • F Distribution • Analysis of Variance (One Way) • Binomial Distribution • Poisson Distribution

■ **Mathematics**

Triangle Solutions • Derivative of a Function • Linear Interpolation • Circle Determined by Three Points



**Reorder Number**  
**00032-90057**

00032-90058 English  
Printed in Canada 6/88

